Corso base di Delphi

Pagina 66
Pagina 65

Corso base di Delphi


Delphi 5 - Corso Base

Introduzione

Obiettivi del corso

Gli obiettivi dei corsi sono:

1. imparare a programmare 

2. imparare a programmare in Pascal e in Object Pascal

3. imparare a programmare in Delphi 

4. affrontare le tematiche della progettazione di una piccola applicazione database

Imparare a programmare significa acquisire una mentalità da programmatore e saper applicare tecniche standard davanti a tipologie specifiche di problemi. Purtroppo non avremo molto tempo per approfondire quanto basta le tecniche di programmazione, ma non tralasceremo di fornire le basi nella convinzione che è inutile saper programmare in Delphi senza conoscere i principi della buona programmazione.

Delphi è un ambiente di programmazione sviluppato da Borland (ora Inprise). Un ambiente di programmazione è un insieme di strumenti integrati (editor, compilatore, linker, debugger più altri eventuali) per la programmazione in un certo linguaggio. Il linguaggio di Delphi è l’Object Pascal, una versione orientata agli oggetti del Pascal

Il Pascal venne inventato da N. Wirth negli anni ’60 come esempio di linguaggio idoneo alla programmazione strutturata. Fin da subito si diffuse universalmente come il linguaggio adatto all’insegnamento, per le sue caratteristiche di semplicità e di limpidezza. Negli anni ’80 la Borland sviluppò un compilatore Pascal (il Turbo Pascal) che in poco tempo ottenne il monopolio del mercato. Da questa posizione di dominio ha cominciato a modificare le caratteristiche del linguaggio rendendolo più adatto all’uso di programmatori professionisti e concorrenziale con linguaggi più blasonati come il C. Object Pascal e Delphi sono gli eredi del TurboPascal.

Imparare a programmare in Delphi significa imparare a programmare in Pascal prima e in Object Pascal poi. In questo corso introduttivo svilupperemo alcune tecniche di programmazione Pascal standard su console prima di passare alla programmazione orientata agli oggetti e allo sviluppo di programmi con interfacce grafiche. L’obiettivo di questa prima fase è cominciare a familiarizzare col linguaggio imparando tecniche di programmazione che saranno utili nello sviluppo di qualsiasi tipo di applicazione. Apprendere queste tecniche aiuterà la formazione di una mentalità da programmatore.

Passando alla fase successiva si prenderà confidenza con i componenti più usati della Visual Component Library (VCL). Delphi utilizza un modello di programmazione chiamato RAD (Rapid Application Development) che consiste nella capacità di disegnare l’interfaccia dell’applicazione trascinando le sue varie componenti (pulsanti, etichette, caselle di modifica testo) direttamente sulla form (la controparte progettuale della finestra). Ad ogni “pezzo” dell’interfaccia grafica corrisponde un componente. Più avanti in questa stessa lezione vedremo alcuni esempi.

Infine nel corso avanzato svilupperemo una piccola applicazione database di esempio. Poiché questo argomento non riguarda il corso introduttivo non c’è bisogno qui di ulteriori approfondimenti

L’interfaccia del programma

Aprendo il programma per la prima volta si vedono 4 finestre

In alto abbiamo la finestra principale che contiene il menu principale, le barre degli strumenti e la component palette. Da menu si può accedere a tutte le funzionalità di Delphi, mentre le barre degli strumenti (configurabili dall’utente) forniscono un accesso rapido alle funzionalità più usate.

La finestra a sinistra è l’Object Inspector che visualizza le proprietà e i gestori d’evento del componente attualmente selezionato.

In centro ci sono due finestre: il Code Editor (nella foto in bianco col titolo “Unit1.pas”) che serve per scrivere il codice e il Form Designer (nella foto quello col titolo “Form1”) che è la superficie su cui trascineremo i componenti dalla Component Palette.

Esempi di programmazione

Iniziamo il corso con alcuni semplici esempi che ci permetteranno di capire le potenzialità della programmazione in Delphi. Partiremo naturalmente col classico “Hello world”, vedremo in seguito un esempio che apre due finestre e proseguiremo con un piccolo programma che fa la somma di due numeri. Infine creeremo una versione per console di quest’ultimo programma. Questo ci introdurrà ai segreti della programmazione console, che approfondiremo nell’ambito delle lezioni successive.

Hello World – versione 1

Si apra una nuova applicazione. Si faccia click sul Form Designer, in modo che sull’Object Inspector siano visualizzate le proprietà della Form. Si selezioni sull’Object Inspector la proprietà Caption e la si modifichi a “Hello world”. Si esegua il programma (F9). Abbiamo generato un programma  con una finestra vuota e col titolo “Hello world”. La finestra è una normalissima finestra Windows che può essere spostata, ridimensionata e ridotta a icona.

Hello World – versione 2

Una possibile variante di questo programma visualizza la scritta “Hello world” nel centro della finestra. Si trascini un componente Label dalla component palette posizionandolo più o meno in centro alla form. Ora si selezioni il componente label (con nome Label1) e si modifichi la sua proprietà Caption a “Hello world”. Se eseguiamo il programma avremo una finestra con una scritta centrata.

Il componente label serve per scrivere del testo statico. Selezionando il componente Label1 ci vengono mostrate nell’Object Inspector le sue proprietà. Vediamone alcune

Caption

Contiene il testo della label. Viene inizializzato col nome di default del componente (Label1, Label2,..)

Left,Top,Width,Height

Queste proprietà impostano la posizione della label all’interno della form e la sua dimensione

Color

Il colore dello sfondo della label

Font

Font è una proprietà composta che ha un menu di sottoproprietà. Se la lista di sottoproprietà è espansa nell’Object Inspector allora sulla sinistra della proprietà font compare un simbolo ‘-‘, altrimenti compare un simbolo ‘+’

Segue una breve descrizione delle sottoproprietà:

	Sottoproprietà
	Significato

	Charset
	Set di caratteri. Da modificare solo se non si lavora con caratteri delle lingue dell’Europa Occidentale

	Color
	Il colore del testo

	Height
	L’altezza del font in pixel

	Name
	Il nome del font (e.g.: Times New Roman)

	Pitch
	Proprietà che permette di forzare la variabilità in larghezza di un font

	Size
	L’altezza del font in punti

	Style
	Altra proprietà composta: si può specificare gli stili Grassetto, Corsivo, Sottolineato, Barrato


Le proprietà interessanti sono Color, Name, Size e Style. In circostanze normali non serve cambiare il valore di default di Charset e Pitch, mentre Height fornisce all’applicazione la possibilità di leggere e scrivere l’altezza del font in pixel invece che in punti.

Se si seleziona la riga principale di Font nell’Object Inspector comparirà il simbolo ‘...’. Cliccandolo si accede al property editor specializzato che visualizza un comune font dialog con cui è più pratico modificare le proprietà del font.

Autosize

Scrivendo testo nella proprietà Caption si noterà che le dimensioni della label cambiano con la lunghezza del testo. La stessa cosa capita se si cambiano le dimensioni del font: il componente label si adeguerà alla nuova altezza e alla nuova larghezza del testo contenuto in Caption. 

Il motivo è che la proprietà Autosize è impostata di default a true. Se Autosize è a false le dimensioni del componente non cambieranno a meno che non le modifichiate da form designer o modificando le proprietà Left, Top, Width, Height

Name

E’ il nome della label, quello con cui vi riferirete alla label nel vostro codice

Hello world – versione 3

Ora aggiungiamo un componente button alla form. Cliccando sul pulsante cambieremo il contenuto della label da “Hello World” a “Ciao mondo”.

Prima di tutto si trascini il componente button sulla form, in una posizione qualsiasi. Quindi si modifichi la sua proprietà Caption a “Cambia lingua”.

N.B: si osservi che anche il componente button ha le proprietà Caption, Left, Top, Width, Height, Font, Name e che non ha Autosize e Color. Si osservi che è possibile cambiare le caratteristiche del font ma non il colore del testo (la modifica non avrà effetto).

Quello che vogliamo ora è che premendo il pulsante cambi la Caption della label. Per far questo selezioniamo Button1, ci spostiamo sulla linguetta Events nell’Object Inspector e facciamo doppio click sulla prima riga, “OnClick”. Succedono 3 cose: del codice viene automaticamente aggiunto nel Code Editor, il Code Editor viene portato davanti alle altre finestre e riceve il focus. Il codice inserito da Delphi sarà:

procedure TForm1.Button1Click(Sender: TObject);

begin

|

end;

dove la sbarra verticale rappresenta il cursore. A questo punto inseriamo il seguente codice:

procedure TForm1.Button1Click(Sender: TObject);

begin

  Label1.Caption := ‘Ciao Mondo’

end;

E’ fatta. Se eseguiamo il programma e poi facciamo click sul pulsante “Cambia lingua” il contenuto della label passa da “Hello World” a “Ciao Mondo”

Se si volesse cambiare lingua tutte le volte che si preme il pulsante (da inglese a italiano e viceversa) il codice da scrivere sarebbe:

procedure TForm1.Button1Click(Sender: TObject);

begin

  if Label1.Caption = ‘Hello world’ then Label1.Caption := ‘Ciao Mondo’

  else Label1.Caption := ‘Hello world’

end;

Così come abbiamo cambiato la proprietà Caption si potevano cambiare le altre proprietà. Per esempio il seguente codice:

procedure TForm1.Button1Click(Sender: TObject);

begin

  if Label1.Caption = ‘Hello world’ then
  begin

    Label1.Caption := ‘Ciao Mondo’;

    Label1.Color := clGreen;

    Label1.Font.Name := ‘Times New Roman’

  else 

  begin

    Label1.Caption := ‘Hello world’;

    Label1.Color := clWindowText;

    Label1.FontName := ‘MS Sans Serif’

  end

end;

Cambia oltre che al testo anche il tipo di font e il colore di sfondo della label

Two forms

Il secondo esempio che proviamo apre due form invece di una form sola.

Dopo avere chiuso l’esempio precedente col comando di menu File|Close all e aver aperto un’applicazione nuova con File|New Application trasciniamo un componente button sulla form vuota e modifichiamo la sua proprietà Caption in “Seconda form”. Quindi apriamo una seconda form col comando File|New form (esiste un pulsante anche nella barra degli strumenti:).

Quello che vogliamo è che all’apertura del programma si veda una form sola ma che premendo il pulsante “Seconda form” compaia una seconda form vuota.

Per fare questo selezioniamo il nostro pulsante, andiamo sulla riga OnClick della sezione Eventi dell’Object Inspector e facciamo doppio click (un modo alternativo è fare doppio click sul pulsante stesso nel Form Designer). Come prima apparirà il Code Editor con il codice:

procedure TForm1.Button1Click(Sender: TObject);

begin

|

end;

Scriviamo il seguente codice:

procedure TForm1.Button1Click(Sender: TObject);

begin

  Form2.Show

end;

Il gioco è fatto. Premendo sul pulsante “Seconda Form” ora apparirà una seconda form. E’ possibile farla scomparire chiudendola nel modo tradizionale e farla riapparire premendo nuovamente il pulsante  

Somma di 2 numeri GUI

Il prossimo esempio è un programmino che chiede in input due numeri interi e restituisce la somma. Per farlo abbiamo bisogno di un nuovo tipo di componente, la casella di modifica testo, Edit Box in inglese o più semplicemente Edit.

Ci servono 3 componenti Edit, 3 componenti Label e 1 pulsante.

La form alla fine deve essere simile a questa:

Finora abbiamo sempre usato i nomi dei componenti affibbiati da Delphi ma d’ora in poi daremo nomi più significativi a tutti i componenti che ci capiterà di usare nel codice. E’ buona norma ridenominare tutti i componenti tranne le label subito dopo averli trascinati sulla form.

Dopo aver trascinato i componenti come mostrato nella figura occorre:

1. Cambiare la Caption della Form a “Operazioni aritmetiche tra due numeri”

2. Cambiare Label1.Caption a “Op. 1”

3. Cambiare Label2.Caption a “Op. 2”

4. Cambiare Edit1.Text a “0”

5. Cambiare Edit1.Name a “Op1Edit”

6. Cambiare Edit2.Text a “0”

7. Cambiare Edit2.Name a “Op2Edit”

8. Cambiare Button1.Caption a “Somma”

9. Cambiare Button1.Name a “SommaBtn”

10. Cambiare Label3.Caption a “Risultato”

11. Svuotare Edit3.Text 

12. Cambiare Edit3.Name a “RisultatoEdit”

Ora abbiamo 1 pulsante (SommaBtn) e  3 edit (Op1Edit, Op2Edit e RisultatoEdit) come parti attive della form.

Si noti che il componente Edit ha le proprietà Left, Top, Width, Height, Color, Font, Name. Non ha una proprietà Caption ma una proprietà equivalente Text. Analogamente a Label ha anche una proprietà Autosize, che non analizzeremo ora.

Per completare il codice bisogna fare doppio click su SommaBtn nel Form Designer in modo che compaia il solito codice di gestione dell’evento OnClick. All’interno del gestore d’evento occorre scrivere il seguente codice:

procedure TForm1.SommaBtnClick(Sender: TObject);

var

  op1, op2: integer;

  risultato: integer;

begin

  op1 := StrToInt(Op1Edit.Text);

  op2 := StrToInt(Op2Edit.Text);

  risultato := op1 + op2

  RisultatoEdit.Text := IntToStr(risultato)

end;

Ora inserendo due numeri nei primi 2 edit comparirà la somma nel terzo edit, quello del risultato. Nelle prime 3 righe del nostro codice dichiariamo le variabili che ci servono (op1, op2, risultato). Il codice vero e proprio è quello che sta tra begin ed end: la prima e la seconda riga trasferiscono le proprietà Text degli Edit di input (stringhe) nelle variabili op1 e op2 (interi). La terza riga esegue l’operazione di somma vera e propria, il cui risultato viene messo nella variabile locale risultato. La quarta e ultima riga trasferisce il valore memorizzato in risultato nella proprietà Text di RisultatoEdit.

Un problema di questo programma è che il contenuto dell’edit RisultatoEdit è modificabile dall’utente, mentre il suo unico scopo dovrebbe essere quello di mostrare il risultato. Edit ci viene incontro con la proprietà ReadOnly che di default è impostata a false ma che se viene impostata a true impedisce all’utente di modificare il testo.

Un altro problema è che se l’utente inserisce caratteri alfabetici invece di testo negli edit di input il programma viene temporaneamente bloccato per mostrare un messaggio d’errore in inglese. Supponiamo che questo comportamento non sia quello desiderato ma che preferiamo piuttosto che il programma visualizzi un risultato nullo in caso di input non valido. Quello che dobbiamo fare è intercettare l’errore e gestirlo nel modo che preferiamo. Nel gergo della programmazione orientata agli oggetti un errore è chiamato “eccezione”.

Per ottenere questo risultato dobbiamo modificare il codice sopra in questo modo

procedure TForm1. SommaBtnClick (Sender: TObject);

var

  op1, op2: integer;

  risultato: integer;

begin

  try

    op1 := StrToInt(Op1Edit.Text);

    op2 := StrToInt(Op2Edit.Text);

    risultato := op1 + op2

    RisultatoEdit.Text := IntToStr(risultato)

  except

  on EConvertError do RisultatoEdit.Text := '0';

  end;

end;

Le quattro righe che prima componevano il corpo della procedura ora sono racchiuse tra le parole try..except. Qualsiasi errore si verifichi qui dentro viene gestito nel codice racchiuso tra le parole riservate except...end. Tale codice dice quale tipo di eccezione gestire (EConvertError) e in tal caso esegue il codice che assegna “0” alla proprietà Text di RisultatoEdit

Somma di 2 numeri DOS

Come ultimo esempio della prima lezione apriamo un’applicazione console. Per farlo selezionare da menu File|New... e nel dialog box che segue fare doppio click su “Console Application” (è la quarta opzione della prima riga).

Il codice che ci troviamo davanti è il seguente

program Project1;

{$APPTYPE CONSOLE}

uses SysUtils;

begin

  // Insert user code here

end.

Salviamo il progetto col nome “Somma” nella cartella “Somma DOS” e modifichiamo il listato in modo che diventi simile al seguente:

program Somma;

{$APPTYPE CONSOLE}

uses

  SysUtils;

var

  risultato: integer;

  op1, op2: integer;

begin

  writeln('SOMMA DI DUE NUMERI');

  writeln;

  write('Dammi il primo numero: ');

  readln(op1);

  write('Dammi il secondo numero: ');

  readln(op2);

  risultato := op1 + op2;

  write('Il risultato e'' ',risultato);

  readln;

end.

Questo programma apre una finestra DOS, scrive “SOMMA DI DUE NUMERI”, chiede due numeri in input e fornisce in output la somma, quindi aspetta che l’utente prema invio prima di uscire.

Si noti che come prima cosa abbiamo dichiarato 3 variabili, op1, op2 e risultato.

Per l’output vengono usate 2 funzioni:

· write(argomento1,...,argomentoN) scrive tutto ciò che c’è tra parentesi tonde sulla console 

· writeln(argomento1,...,argomentoN) fa la stessa cosa di write, dopodiché va a capo

Per l’input viene usata la funzione:

· readln(variabile1,...variabileN) imposta il contenuto di n variabili a quello che scrive l’utente sulla console, terminando di leggere dopo aver riempito tutte le variabili e dopo aver ricevuto INVIO.

Si osservi il codice tra le parole begin e end: le prime 2 righe scrivono il titolo sulla console (la seconda scrive una riga vuota). Le successive 4 righe chiedono in input i valori dei due operandi e li memorizzano nelle variabili op1 e op2 (si noti l’uso di write invece che di writeln). La quinta riga esegue l’operazione vera e propria memorizzando il  risultato nella variabile risultato. La riga successiva scrive il risultato. Infine l’ultima riga attende che l’utente prema INVIO prima di uscire.

Questo era solo un assaggio. Nelle prossime lezioni si spiegherà meglio tutte le caratteristiche del linguaggio, prese una a una. Prima di iniziare a spiegare le caratteristiche del linguaggio Pascal vediamo come è fatto un computer moderno

Architettura di un calcolatore

Gli elementi essenziali di un computer moderno sono la C.P.U., la memoria centrale e i dispositivi di I/O.

La C.P.U. (Central Processing Unit) o processore è il cuore del computer ed è incaricato di eseguire le istruzioni nei programmi. I processori più usati sugli attuali personal computer sono il Pentium (II, III o IV) sui PC-compatibili e il PowerPC sugli Apple Macintosh. La velocità del processore si calcola in Mega Hertz (Mhz) ed esprime quanti milioni di operazioni macchina al secondo è capace di eseguire. Velocità tipiche degli attuali personal vanno dai 200Mhz ai 1100 Mhz (da 200 milioni fino a 1 miliardo e 100 milioni di operazioni macchina al secondo).

La memoria centrale contiene sia i dati dei programmi sia i programmi veri e propri. La memoria è composta da una sequenza lineare di celle di memoria, chiamate byte. Un byte è a sua volta costituito da 8 cifre binarie (bit, per binary digit). Ogni cifra binaria può assumere i valori 0 o 1. Combinando insieme 8 cifre binarie è possibile codificare 28 = 256 valori. Un byte può perciò contenere un numero che può assumere 256 valori differenti (da 0 a 255). Ogni carattere alfanumerico viene codificato all’interno di un byte, mentre per memorizzare numeri interi si ricorre di solito all’uso combinato di 4 byte (32 bit), permettendo così di memorizzare fino a 232 = 4.294.967.296 valori differenti. I multipli del byte sono il kilobyte (kB) (1024 byte), il megabyte (MB) (1024 kB ~ 1 milione di byte), il gigabyte (GB) (1204 MB ~ 1 miliardo di byte) e il terabyte (TB) (1024 GB ~ 1.000 miliardi di byte). Le memorie dei personal attuali vanno dai 32 MB ai 256 MB (ossia permettono di memorizzare 32-256 milioni di caratteri alfanumerici).

La C.P.U. e la memoria centrale costituiscono il computer vero e proprio. Il terzo elemento che costituisce un computer completo sono i dispositivi (device) di Input / Output (I/O). Esempi di questi dispositivi sono la tastiera, il mouse, il monitor, il floppy disk, l’hard disk, il CD-ROM. Una famiglia particolare di dispositivi di I/O è costituita dalle memorie di massa che operano come dispositivi di immagazzinamento dati a lungo termine (mentre la memoria centrale si svuota non appena si spegne il computer). Le memorie di massa sono solitamente più capienti e più lente della memoria centrale. Il dispositivo di memoria di massa più importante è l’hard disk o disco rigido. La capienza degli hard disk moderni si misura in GB e le misure vanno dai 2-4 GB fino ai 60 GB.

L’architettura di un computer moderno può essere schematizzata dalla figura seguente:

Questo modello, in cui i dati e i programmi risiedono nella memoria, è stato concettualizzato alla fine degli anni ’40 e prende il nome di macchina di Von Neumann dallo studioso che l’ha teorizzato. La Macchina di Von Neumann ha un funzionamento tipicamente sequenziale ed è alla base del funzionamento di tutti i computer moderni (con l’eccezione di alcune architetture hardware sperimentali, come i transputer).

Esiste un linguaggio speciale, il linguaggio macchina, che è l’unico ad essere eseguito dalla C.P.U. In una macchina di Von Neumann un processore passa tutto il suo tempo a leggere istruzioni in L/M (linguaggio macchina) dalla memoria, a eseguirle e a recuperare la prossima istruzione da eseguire. L’istruzione successiva sarà quella immediatamente successiva in memoria se l’istruzione precedente non è un salto, altrimenti quella che si trova all’indirizzo specificato dall’istruzione di salto. Un’istruzione in linguaggio macchina può muovere dati, eseguire operazioni aritmetiche elementari, coinvolgere un dispositivo di I/O, eseguire un salto sotto una certa condizione o eseguire un salto incondizionato. Da questo punto di vista è in tutto e per tutto simile a un linguaggio ad alto livello, tranne che per il fatto che le operazioni che può eseguire sui dati sono estremamente semplici (nei processori più vecchi non esistevano neppure la moltiplicazione e la divisione, che andavano simulate tramite piccoli programmi).

Poiché il linguaggio macchina è una sequenza di 0 e 1, esiste una notazione che fornisce nomi mnemonici alle istruzioni L/M e programmi che convertono file sorgente scritti in questo linguaggio nel linguaggio macchina vero e proprio. Il linguaggio sorgente è chiamato Assembly mentre il programma che esegue la conversione è chiamato Assembler e la conversione stessa assemblare o assemblaggio. Anche se impropriamente si è ormai diffusa la consuetudine di riferirsi al linguaggio col nome di Assembler.

Esiste una catalogazione di tutti i linguaggi di programmazione in linguaggi di basso, medio e alto livello, a seconda di quanto siano vicini al modo di operare della macchina o di quanto piuttosto adottino un’impostazione più vicina al modo di pensare umano. In particolare il linguaggio macchina è di basso livello, perché vicino alla macchina, l’Assembler di medio livello e tutti gli altri linguaggi di alto livello. I linguaggi di alto livello si dividono in compilatori e interpreti. I compilatori sono quei linguaggi che vengono tradotti direttamente in L/M, mentre gli interpreti traducono i sorgenti in un codice intermedio che necessita ancora di un programma che esegua la conversione al linguaggio macchina a tempo di esecuzione. Mentre il Basic era originariamente un interprete, oggi esistono compilatori di Basic (come Microsoft Visual Basic). Linguaggi che sono nati e restano compilatori sono il Pascal, il C, il Fortran, il Cobol e molti altri.

Esistono alcune famiglie di linguaggi ad alto livello: procedurali (Pascal, C, Basic, Fortran, Cobol), funzionali (Lisp), logici (Prolog). Ciascuna di queste famiglie rappresenta una differente filosofia di approccio alla programmazione. E’ inutile approfondire ora le metodologie di ogni famiglia, quel che importa è sapere che il Pascal è un linguaggio procedurale. La programmazione procedurale è quella classica: in un linguaggio procedurale i programmi sono costituiti da liste sequenziali di istruzioni che il computer dovrà eseguire una alla volta. Un’evoluzione della metodologia procedurale è la programmazione orientata agli oggetti (OOP – Object Oriented Programming). Delphi utilizza Object Pascal, che è un’estensione OOP del Pascal originale. Vedremo durante il corso cos’è la OOP. Altri esempi di linguaggi OOP sono Java, C++ e SmallTalk. 

Pascal

Dopo questa premessa teorica iniziamo a parlare di Pascal. Prima di tutto cominceremo col parlare delle caratteristiche del linguaggio originale e a tal proposito vedremo esempi che fanno uso di un’interfaccia a linea di comando, mentre torneremo alle interfacce  grafiche sotto Windows quando parleremo di programmazione orientata agli oggetti.

Struttura di un programma Pascal

La struttura di un programma Pascal è essenzialmente la seguente:

Program <nome programma>;

const


...

type


...

var


...

begin


...

end.

Possono esserci altri elementi ma queste sono le sezioni più importanti di un programma.

La prima riga è l’intestazione del programma. Contiene la parola riservata program, il nome del programma e un punto e virgola. Nel Pascal originale poteva seguire la lista dei file esterni tra parentesi; in Delphi e in Turbo Pascal la lista dei file esterni se esiste viene ignorata.

All’intestazione del programma seguono le sezioni di dichiarazione delle costanti, dei tipi di dati e delle variabili. L’inizio di una nuova sezione termina la precedente. Quest’ordine non è obbligatorio e anzi possono esserci più sezioni di uno stesso tipo. L’unica limitazione è che non è possibile utilizzare una costante, un tipo di dati o una variabile senza prima averli dichiarati; con un po’ di esperienza si scoprirà che la disposizione indicata sopra è quella più naturale.

Dopo le dichiarazioni segue il corpo del programma, delimitato dalle parole riservate begin e end e terminato col simbolo ‘.’. All’interno del corpo sta il programma principale. 

Ecco un esempio di programma:

program AreaPerimetroCerchio;

{$APPTYPE CONSOLE}

uses

  SysUtils;

const

  Pi = 3.14;

type

  NumeroDecimale = real;

var

  Area, Perimetro: NumeroDecimale;

  Raggio: NumeroDecimale;

begin

  { titolo }

  writeln('CALCOLO DELL''AREA E DEL PERIMETRO DI UN CERCHIO');

  writeln;

  { input }

  write('Inserisci il raggio del cerchio: ');

  readln(Raggio);

  { calcolo }

  Area := Pi * Raggio * Raggio; // Area := pi * r ^ 2

  Perimetro := 2 * Pi * Raggio; // Perimetro := 2 * pi * r

  { output }

  writeln('L''area del cerchio e'' ',Area);

  writeln('Il perimetro del cerchio e'' ',Perimetro);

  readln;

end.

Analizziamo questo listato riga per riga. 

Le prime 4 righe sono state inserite da Delphi. La prima riga ci dice che il nome del programma è CalcoloAreaPerimetro. In effetti il programma chiede il raggio di un cerchio e restituisce in output l’area e il perimetro. La seconda riga è una direttiva di compilazione che indica al compilatore che questa applicazione è di tipo console. La terza e la quarta riga (uses SysUtils;), sempre inserite da Delphi, permettono al nostro programma di utilizzare alcune funzioni di sistema definite nella unit SysUtils. Non serve capire tutto ora, queste cose si capiranno meglio più avanti.

Proseguiamo con l’analisi del codice. Nella sezione delle dichiarazioni definiamo una costante pi che contiene la costante matematica π (pi greco). E’ noto che l’area di un cerchio è data dalla formula πr2 e che il perimetro è dato dalla formula 2πr,  dove r è il raggio del cerchio. Poi definiamo per puri scopi didattici un tipo di dati, NumeroDecimale, che sarà il tipo delle variabili che dichiareremo. NumeroDecimale non è altro che un real cambiato di nome, ma il suo utilizzo ci permette di fornire un esempio di dichiarazione di tipo. Infie dichiariamo 3 variabili: Area, Perimetro, Raggio, tutte e tre di tipo NumeroDecimale.

Finita la sezione delle dichiarazioni inizia il codice vero e proprio, delimitato dalle parole begin e end. La prima riga nel codice ({titolo}) è un commento inserito per maggiore chiarezza. Apriamo una breve parentesi sui commenti.

I commenti si possono inserire nei listati Pascal in 3 modi:

· delimitati da parentesi graffe 

· delimitati da (* e *)  

· iniziati da // e terminati dalla fine della riga 

Esempi:

...

{ questo è un commento }

(* questo è un commento *)

// questo è un commento

Mentre le prime due sintassi sono equivalenti e appartengono al Pascal standard, l’ultima sintassi è stata copiata dal C++ (i commenti scritti così sono chiamati C-like).

Anche se le prime due sintassi sono equivalenti, vengono usate di solito dai programmatori in maniera distinta. E’ ad esempio una prassi comune usare le graffe per i commenti normali e riservare l’uso di (* ... *) al codice commentato per debug. 

L’uso del terzo tipo di commenti è una questione di puro gusto personale. L’unica controindicazione è quella che usandolo si scrive codice non compatibile con altri compilatori Pascal e con Delphi 1. Di solito è molto utile per commentare singole righe in fase di debug, ma molti programmatori lo usano anche per i commenti comuni.

Se ora rianalizziamo la seconda riga ({$ APPTYPE CONSOLE}) vediamo che in realtà ci troviamo di fronte a un commento un po’ anomalo. In realtà la sintassi {$ ... } o anche (*$ ... *) serve per specificare le direttive di compilazione. Le direttive di compilazione sono comandi per il compilatore che esprimono una preferenza del programmatore sul modo in cui verrà generato il codice. La direttiva usata qui ad esempio è la APPTYPE che ha due possibili valori: GUI (applicazione normale Windows con interfaccia grafica) e CONSOLE (applicazione console con interfaccia a linea di comando). 

Andiamo avanti. Le prossime due righe scrivono il titolo del programma più una riga vuota. Poi seguono due righe che chiedono all’utente il raggio e lo memorizzano nella variabile Raggio. Le due righe che seguono effettuano i calcoli veri e propri:

  { calcolo }

  Area := Pi * Raggio * Raggio; // Area := pi * r ^ 2
  Perimetro := 2 * Pi * Raggio; // Perimetro := 2 * pi * r

Si noti che il simbolo ‘:=’ esegue un assegnamento di un valore (che è il risultato dell’espressione di destra) a una variabile (che è Area nella prima riga e Perimetro nella seconda). Mentre in altri linguaggi l’assegnamento si effettua col simbolo ‘=’, in Pascal si utilizza ‘:=’ per l’assegnamento e ‘=’ per il confronto. Il codice si commenta da sé (il * è il simbolo di moltiplicazione).

Terminata l’esecuzione di queste due righe le variabili Area e Perimetro contengono i valori da mostrare all’utente. Questo compito viene svolto dalle prossime due righe del programma

  writeln('L''area del cerchio e'' ',Area);

  writeln('Il perimetro del cerchio e'' ',Perimetro);

Non resta che chiudere il programma con la solita readln per impedire che il programma richiamato da IDE si chiuda immediatamente.

Notiamo alcune cose sulla sintassi utilizzata:

· quasi tutte le righe terminano con ‘;’. Il punto e virgola è un separatore in Pascal: separa l’intestazione del programma dal corpo, una sezione di dichiarazioni (const, var, type) dalla successiva, una dichiarazione dalla successiva (si vedano le due righe di dichiarazione variabili), un’istruzione dalla successiva (si veda nel corpo della funzione). Non serve metterlo prima di end per cui il ‘;’ alla fine dell’ultima istruzione nel corpo (readln) è superfluo, ma è comunque una buona idea metterlo per evitare di dimenticarselo quando in un secondo tempo si espanderà il codice

· nella definizione delle costanti si usa il simbolo ‘=’ invece che ‘:=’ perché viene considerata un’equivalenza matematica. Mentre a una variabile si assegna un valore che può cambiare nel corso del programma una costante ha un valore definito all’inizio del programma e immutabile

· anche nella definizione dei tipi si usa il simbolo ‘=’

· nella dichiarazione delle variabili l’identificatore (cioè il nome) della variabile sta a sinistra del simbolo ‘:’, mentre a destra poniamo il tipo della variabile. I tipi di variabile che abbiamo visto finora sono i tipi predefiniti ‘integer’ (numero intero) e ‘real’ (numero con cifre decimali) e abbiamo visto che possiamo definire tipi personalizzati. Più avanti vedremo altri tipi predefiniti tra cui ‘boolean’ (valore di verità), ‘char’ (carattere) e ‘string’ (stringa o sequenza di caratteri)

Provando il programma vediamo che il risultato è nel formato <numero>E<segno><cifre>, p.e. 2.82600000E+001. Questa è la notazione scientifica. A destra del simbolo E c’è l’esponente, a sinistra la mantissa. Il numero sopra è da leggere come: 2.826 * 101 = 28.26 (è l’area del cerchio di raggio 3).

Costrutti del linguaggio

Passiamo ora ad esaminare quali sono i principali costrutti del linguaggio. Esistono sostanzialmente due grandi famiglie che assorbono quasi tutti i costrutti: quelli condizionali, che servono per eseguire codice differente a seconda del valore di una variabile e quelli ciclici che servono per ripetere più volte l’esecuzione di un blocco di codice. Esiste poi un’istruzione di salto (goto) per saltare incondizionatamente da una parte all’altra del codice e una sintassi che esprime un’istruzione composta da una serie di istruzioni. Partiamo da quest’ultima

Istruzioni composte

Molti tra i costrutti condizionali e di ciclo che vedremo tra breve prevedono nella loro sintassi che si utilizzi al loro interno singole istruzioni. Poiché invece in tali circostanze si desidera spesso inserire più di un’istruzione, è possibile a tal fine specificare un’istruzione composta, ossia un’istruzione delimitata dalle parole chiave begin e end che contiene molte istruzioni ma che a livello sintattico vale come una. La sintassi è la seguente:

begin

  statement1;

  statement2;

  ...

  statementN

end;

Ogni statement è separato dal prossimo da un punto e virgola; solo il punto e virgola dell’ultimo statement è facoltativo (anche se è una buona idea metterlo)

Costrutti condizionali

I costrutti condizionali sono 2:

· if..then..[else ...]

· case ... of...

If..then

Sintassi
Ci sono due sintassi alternative:

if expression then statement
if expression then statement1 else statement2
La prima esegue statement se l’espressione booleana expression vale true. La seconda esegue statement1 se l’espressione booleana expression vale true, altrimenti esegue statement2.

La sintassi richiede che statement, statement1 e statement2 siano singole istruzioni. Per inserire più istruzioni occorre usare un’istruzione composta

Esempi

if J = 0 then

  Exit

else

  Result := I/J;

if Label1.Caption = ‘Hello world’ then Label1.Caption := ‘Ciao Mondo’

else Label1.Caption := ‘Hello world’;

if J <> 0 then

begin

  Result := I/J;

  Count := Count + 1;

end

else if Count = Last then

  Done := True

else

  Exit;

Note
(1) Si noti che mettere ‘;’ prima dell’else provoca un errore di compilazione

(2) Si consideri la seguente istruzione:

if expression1 then if expression2 then statement1 else statement2;

Come va interpretata? Effettivamente ci sono due possibili interpretazioni:

if expression1 then [ if expression2 then statement1 else statement2 ];

if expression1 then [ if expression2 then statement1 ] else statement2;

Cioè o c’è un’istruzione if..then..else che è lo statement di una if..then oppure c’è una if..then che è lo statement1 di una if..then..else. La soluzione corretta è la prima. La regola è che l’else si associa sempre all’if più vicino. 

(3) Molto spesso si utilizza una serie di if..then..else innestati per verificare condizioni alternative. Per esempio:

if a > 0 then writeln(‘il valore e’’ positivo’)

else if a < 0 then writeln(‘il valore e’’ negativo’)

else writeln(‘il valore e’’ 0’)

Case

Sintassi

La sintassi è la seguente.

case selectorExpression of
  caseList1: statement1;

  ...

  caseListn: statementn;

[else
  statement; ]

end

dove selectorExpression è un’espressione ordinale e i vari caseList sono liste di espressioni costanti che possono includere anche sintassi subrange (valoreInziale..valoreFinale). Ogni caso prevede una singola istruzione; ovviamente quando ne serve più di una si ricorre alle istruzioni composte.

L’istruzione else è facoltativa e viene eseguita quando tutti gli altri casi sono falsi

Esempi

case I of
  1..5: Caption := 'Low';

  6..9: Caption := 'High';

  0, 10..99: Caption := 'Out of range';

else

  Caption := '';

end;

case ch of
  'a'..'z': writeln('Il carattere e'' minuscolo');

  'A'..'Z': writeln('Il carattere e'' maiuscolo');

  '0'..'9': writeln('Il carattere e'' una cifra');

else writeln('Il carattere e'' un simbolo non alfanumerico');

end;

case Selection of
  Done: Form1.Close;

  Compute: CalculateTotal(UnitCost, Quantity);

else

  Beep;

end;
Note

Le liste di casi non devono sovrapporsi. 

Costrutti ciclici

I costrutti ciclici sono 3:

· while

· repeat

· for

While

Sintassi

La sintassi è la seguente:

while expression do statement
Lo statement viene eseguito fino a che l’espressione è vera. Se l’espressione è falsa fin dall’inizio del ciclo lo statement non verrà mai eseguito. Dopo ogni iterazione l’espressione viene rivalutata.

Per iterare più istruzioni usare un’istruzione composta.

Esempi

while Data[I] <> X do I := I + 1;

while not Eof(InputFile) do
begin

  Readln(InputFile, Line);

  Process(Line);

end;

Repeat

Sintassi

La sintassi è la seguente:

repeat statement1; ...; statementn; until expression
Gli statement all’interno del ciclo vengono eseguiti almeno una volta. Alla fine del ciclo se l’espressione è ancora falsa si itera nuovamente. Le differenze con il while perciò sono 3:

· nel corpo trovano spazio più statement, per cui non serve usare un’istruzione composta per iterare più istruzioni

· l’espressione deve essere falsa e non vera perché l’iterazione continui

· il ciclo viene iterato almeno una volta

Esempi

repeat

  Write('Enter a value (0..9): ');

  Readln(I);

until (I >= 0) and (I <= 9);

Note

Pensare alla negazione dell’espressione comporta una difficoltà aggiunta. Per evitarla considerare che l’espressione che compare dopo l’until è quella che deve verificarsi per proseguire nel programma. Nell’esempio sopra se (I  >= 0) e (I <= 9) allora si esce dal ciclo, altrimenti inizia una nuova iterazione.

For

Sintassi

Ci sono due sintassi alternative

for counter := initialValue to finalValue do statement
for counter := initialValue downto finalValue do statement
dove counter è una variabile di tipo ordinale, initialValue e finalValue sono espressioni di un tipo compatibile con counter e statement è un’istruzione semplice o composta.

All’entrata del ciclo counter viene inizializzato a initialValue. Quindi viene eseguita l’istruzione tante volte quanti sono  i valori da initialValue a finalValue. A ogni iterazione la variabile contatore viene incrementata di 1 nella prima sintassi, decrementata di 1 nella seconda sintassi. Nel primo caso initialValue sarà maggiore di finalValue, nel secondo caso finalValue sarà maggiore di initialValue.

Esempi

for i := 1 to 5 do writeln('Iterazione ',i);

for i := 5 downto 1 do writeln('Iterazione ',i);

for i := 1 to Length(s) do
  if s[i] = ch then
  begin

    pos := i;

    break;

  end;

Note

(1) Il ciclo for può essere simulato da un ciclo while con questo codice:

counter := initialValue;

while counter <= finalValue do
begin

  statement;

  counter := counter + 1;

end;

C’è però una differenza sostanziale. Col ciclo while l’espressione finalValue viene rivalutata a ogni iterazione, mentre col ciclo for viene valutata una sola volta prima di entrare nel ciclo

(2) In altri linguaggi (per esempio il Basic) è possibile far fare alla variabile contatore del ciclo for dei salti, ossia incrementarla / decrementarla di un valore diverso da 1 a ogni iterazione. Questo non è possibile in Pascal, per cui è compito del programmatore simulare questo salto. Per esempio se voglio stampare i numeri da 10 a 50 con passo 10 in Basic potemmo scrivere:

for I = 10 to 50 step 10

  print I

endfor
Per fare una cosa del genere in Pascal dovremo scrivere:

for i := 1 to 5 do writeln(i*10);

(3) Non è possibile modificare una variabile di ciclo for all’interno del ciclo: facendolo si ottiene un errore di compilazione.

(4) Se si prova a leggere la variabile contatore oltre la fine naturale del ciclo un warning avverte che la variabile non è definita. Comunque se si prova a eseguire il codice con Delphi 5 si scopre che la variabile ha valore finalValue + 1. Mai e poi mai utilizzare questo valore: è puramente incidentale e potrebbe cambiare nella prossima versione di Delphi. La documentazione avverte che è possibile leggere con sicurezza il valore della variabile contatore dopo la terminazione del ciclo solo se si è verificato un break o un goto dall’interno verso l’esterno.

(5) E’ una convenzione diffusa utilizzare il nome i per la variabile contatore del primo ciclo for, il nome j per un eventuale ciclo for innestato e il nome k per un terzo ciclo for innestato

(6) La variabile di ciclo deve essere locale, ossia definita nella procedura in cui è utilizzato il ciclo for, e non deve essere qualificata

Break, continue

Esistono due istruzioni speciali che possono essere utilizzate all’interno dei cicli for, while e repeat. Queste istruzioni sono break e continue. Break blocca il ciclo più interno: l’esecuzione riprenderà oltre la fine del ciclo. Continue salta alla prossima iterazione del ciclo più interno.

Break e continue sono avversati da una certa accademia che ne vuole l’utilizzo contrario ai dettami della buona programmazione. Anche se in realtà ci sono circostanze in cui l’utilizzo accurato di queste istruzioni può aumentare la leggibilità del codice, un abuso di tali istruzioni può facilmente portare a scrivere codice illeggibile. In particolare sono pochissime le circostanze in cui sarà veramente necessario utilizzare continue, per cui un programmatore principiante può e dovrebbe dimenticarsene. 

Nel caso del break c’è invece una circostanza molto comune che ne rende necessario l’uso ed è quando si esegue una ricerca. Come esempio fornisco qui il listato di un programma che ricerca un carattere in una stringa e ne trova la sua prima occorrenza.
program RicercaCarattere;

{$APPTYPE CONSOLE}

uses

  SysUtils;

var

  s: string;

  i, pos: integer;

  ch: char;

begin

  writeln('RICERCA DI UN CARATTERE IN UNA STRINGA');

  writeln;

  write('Dammi la stringa in cui cercare: ');

  readln(s);

  write('Dammi il carattere da cercare: ');

  readln(ch);

  pos := -1;

  for i := 1 to Length(s) do
    if s[i] = ch then
    begin

      pos := i;

      break;

    end;

  if pos > 0 then
    writeln('Il carattere si trova nella posizione ',pos)

  else

    writeln('Il carattere non si trova nella stringa');

  readln;

end.  

Commentiamo riga per riga il listato. Le variabili utilizzate sono s (la stringa), ch (il carattere da ricercare), i (variabile di ciclo) e pos, che terrà la posizione del carattere all’interno della stringa.

Dopo aver scritto il titolo del programma e aver chiesto all’utente il valore di s e di ch, la variabile pos viene inizializzata al valore –1, ossia a un valore che non potrà assumere all’interno del ciclo. Quindi viene eseguito un ciclo su ogni carattere della stringa. Length(s) è una funzione che restituisce la lunghezza della stringa. La variabile i sarà perciò un indice all’i-esimo carattere della stringa. All’interno del ciclo si guarda se l’i-esimo carattere della stringa (s[i]) è identico al carattere ch. Se è così pos viene posto a i e il ciclo viene terminato con un’istruzione break. 

All’uscita dal ciclo si hanno due casi: o il carattere non era presente nella stringa, nel qual caso pos avrà ancora il valore –1, o il carattere era presente nella stringa, nel qual caso pos avrà il valore dell’indice del primo carattere della stringa identico al carattere cercato.

E’ impensabile pensare di eliminare break da questo codice per una pura questione di stile: in tal caso non ci sarebbe modo di impedire al programma di fare una scansione su tutta la stringa. Si pensi per esempio a un caso in cui si debba eseguire una ricerca su un file lungo molti MB. L’unica soluzione alternativa al break è quella di utilizzare un ciclo while che abbia come condizione di iterazione s[i] <> ch. Un ciclo for è però più leggibile in casi semplici come questi, perché immediatamente si vede dalla sua intestazione su cosa viene fatta l’iterazione, ed è anche leggermente più efficiente, perché la condizione di fine ciclo viene valutata una sola volta.

Ecco lo stesso programma che fa uso di un ciclo while:

program RicercaCarattereWhile;

{$APPTYPE CONSOLE}

uses

  SysUtils;

var

  s: string;

  i: integer;

  ch: char;

begin

  writeln('RICERCA DI UN CARATTERE IN UNA STRINGA');

  writeln;

  write('Dammi la stringa in cui cercare: ');

  readln(s);

  write('Dammi il carattere da cercare: ');

  readln(ch);

  i := 1;

  while (i < Length(s)) and (s[i] <> ch) do i := i + 1;

  if s[i] = ch then
    writeln('Il carattere si trova nella posizione ',i)

  else

    writeln('Il carattere non si trova nella stringa');

  readln;

end.

Anche se questo listato può sembrare equivalente al precedente o addirittura migliore (non viene usata la variabile pos, il ciclo sta su una riga sola) comporta uno svantaggio: la lunghezza della stringa viene valutata a ogni iterazione. Anche se in questo caso il prezzo da pagare è basso, perché per le AnsiString il recupero dell’informazione relativa alla lunghezza della stringa è rapido, ci sono delle circostanze analoghe in cui la valutazione di fine ciclo è un’operazione costosa, per cui se possibile è meglio farla una volta sola invece che a ogni iterazione. 

A questa argomentazione si potrebbe giustamente obiettare che basta utilizzare una variabile aggiuntiva in cui memorizzare il valore di fine ciclo prima dell’inizio del while; in tal caso però si introdurrebbe una variabile al solo scopo di utilizzare una tecnica di programmazione che non è la più adatta alla circostanza e che si rivela perciò meno leggibile dell’altra. Perciò la regola è: utilizzare un ciclo for tutte le volte che si conosce fin dall’inizio qual è il valore finale dell’iterazione, utilizzare un ciclo while quando questa condizione non è nota ma è necessario eseguire una valutazione a ogni iterazione. In un caso come quello descritto sopra l’uso del break non degrada ma anzi migliora la leggibilità.

Costrutti di salto 

L’unica istruzione di salto esplicito è il goto.

goto

Sintassi

La sintassi è la seguente

goto label
La label può essere un identificatore o un numero da 0 a 9999. In entrambi i casi va prima dichiarata come parte delle dichiarazioni locali nel blocco con una riga del genere:

label label1, ..., labeln;

La parte di codice a cui saltare va indicata con la seguente sintassi:

label:
 ...

Esempi

Un esempio da evitare in ogni modo è

label StartHere;

...

StartHere: Beep;

goto StartHere;

Infatti questo codice genera un loop infinito. Ecco una possibile circostanza in cui si può pensare di utilizzare il goto:

procedure FindFirstAnswer;

var X, Y, Z, Count: Integer;

label FoundAnAnswer;

begin

  Count := SomeConstant;

  for X := 1 to Count do
    for Y := 1 to Count do
      for Z := 1 to Count do
        if ... { some condition holds on X, Y, and Z } then
          goto FoundAnAnswer;

  ... {code to execute if no answer is found }

  Exit;

  FoundAnAnswer:

    ... { code to execute when an answer is found }

end;

Note

(1) Non è possibile saltare al di fuori di una procedura o funzione con goto

(2) Mai saltare all’interno di un ciclo: i risultati sono imprevedibili

(3) Evitare l’utilizzo del goto, quale che sia la circostanza: facendolo violerete i principi più elementari della programmazione strutturata. Nei vecchi tempi l’utilizzo del goto era tollerato nella gestione degli errori. Oggi la gestione delle eccezioni presente nei linguaggi OOP rende questa necessità superata

Operatori

Gli operatori si comportano come funzioni predefinite nel linguaggio con una sintassi speciale. Esistono operatori binari che hanno due operandi e restituiscono un valore (p.e. x + y) e operatori unari, con un solo operando (p.e. not a).

Le principali famiglie di operatori sono:

· operatori aritmetici

· operatori logici o booleani

· operatori stringa

· operatori di confronto

· operatori su bit

· operatori su puntatori

· operatori sugli insiemi

· operatori di classe

Per il momento ci limiteremo allo studio delle prime 4 categorie, mentre vedremo l’utilizzo di altri operatori man mano che ci serviranno.

Aritmetici

Gli operatori aritmetici sono:

	Operatore
	Operazione
	Tipi operandi
	Tipo risultato
	Esempio

	+
	Somma
	integer, real
	integer, real
	X + Y

	–
	Sottrazione
	integer, real
	integer, real
	Result - 1

	*
	Moltiplicazione
	integer, real
	integer, real
	P * InterestRate

	/
	Divisione reale
	integer, real
	real
	X / 2

	div
	Divisione intera
	integer
	integer
	Total div UnitSize

	mod
	Resto
	integer
	integer
	Y mod 6


In più si possono considerare operatori aritmetici unari il + e il – che si mette davanti a numeri o a variabili di tipo numerale

	Operatore
	Operazione
	Tipi operandi
	Tipo risultato
	Esempio

	+ (unary)
	Identità segno
	integer, real
	integer, real
	+7

	– (unary)
	Negazione segno
	integer, real
	integer, real
	-X


Note

· Il risultato di / è sempre Extended (un tipo reale); per tutti gli altri operatori il risultato è integer se gli operatori sono entrambi integer, int64 se almeno uno dei due operandi è int64 (e l’altro non è reale) o Extended se almeno uno dei due operandi è reale

· Div esegue la divisione intera, mentre mod restituisce il resto della divisione intera

· Un’eccezione viene sollevata se si esegue una divisione per zero

Logici o Booleani

Gli operatori logici o booleani sono:
	Operatore
	Operazione
	Tipi operandi
	Tipo risultato
	Esempio

	Not
	negazione
	Boolean
	Boolean
	not (C in MySet)

	And
	congiunzione
	Boolean
	Boolean
	Done and (Total > 0)

	Or
	Disgiunzione inclusiva
	Boolean
	Boolean
	A or B

	Xor
	Disgiunzione esclusiva
	Boolean
	Boolean
	A xor B


Intervengono nelle espressioni booleane, ad esempio nell’espressione di valutazione dei cicli while e repeat e nel costrutto if..then.

La congiunzione è vera solo se sono veri entrambi gli operandi, la disgiunzione è vera se è vero un operando o l’altro. Se sono veri entrambi gli operandi la disgiunzione inclusiva è vera mentre quella esclusiva risulta falsa. La negazione è un operatore unario che nega il valore di verità del suo operando. 

Valutare il comportamento di un’espressione logica (booleana) complessa non è sempre un compito banale. Un’espressione logica complessa si compone di espressioni elementari

Per esempio:

(a>=0) or ( (b>=5) and (b<=7) )

Se in questo caso può essere relativamente semplice intuire quando l’espressione assume valore true (cioè quando almeno una delle seguenti condizioni è vera: a è positivo o nullo o b è incluso tra 5 e 7 estremi compresi) il livello di comprensione scende man mano cresce la complessità dell’espressione:

((a<0) and ((b<5) or (b>7))) xor ((a>=0) and ((b>=5) or (b<=7)))

L’espressione sopra è vera in due casi: se a è negativo e b è fuori dal range 5..7 o se a è positivo o nullo e b rientra nel range 5..7. Inoltre utilizzare xor o or in tal caso è completamente equivalente perché le due espressioni non potranno mai essere entrambe vere.

Com’è possibile districarsi di fronte ad espressioni logiche complesse? Quando si ha esperienza la lettura di un’espressione logica diventa più semplice. Ma per chi è agli inizi e anche in certi casi per gli esperti sarebbe di conforto avere un metodo formale con cui poter interpretare rigorosamente le espressioni. Nel prossimo sottoparagrafo forniremo un metodo formale applicabile in certe circostanze.

Espressioni logiche complesse 

Il funzionamento degli operatori logici può essere analizzato attraverso le tabelle di verità. Nella tabella seguente 0 rappresenta il valore di verità false, 1 rappresenta il valore di verità true:

	a
	b
	a AND b
	A OR b
	a XOR b
	NOT a

	0
	0
	0
	0
	0
	1

	0
	1
	0
	1
	1
	1

	1
	0
	0
	1
	1
	0

	1
	1
	1
	1
	0
	0


Le prime due colonne rappresentano tutti i possibili valori degli operandi booleani. Il risultato di a AND b è 1 solo quando sia a che b sono 1, a OR b è 1 quando almeno uno tra a e b è 1, a XOR b è 1 quando o a o b sono 1 ma non entrambi. NOT è un operatore unario: nella tabella NOT nega il valore dell’operando a.

Le tabelle di verità possono essere usate anche per espressioni più complesse, analizzando le parti che le compongono e risalendo fino alla formula completa. Per esempio nell’espressione precedente:

((a<0) and ((b<5) or (b>7))) xor ((a>=0) and ((b>=5) or (b<=7)))

definiamo A = (a >= 0) e B = (b in 5..7)

L’espressione riscritta diventa:

((not a) and (not b)) xor (a and b)

	a
	b
	not a
	Not b
	((not a) and (not b))
	a and b
	((not a) and (not b)) xor (a and b)



	0
	0
	1
	1
	1
	0
	1

	0
	1
	1
	0
	0
	0
	0

	1
	0
	0
	1
	0
	0
	0

	1
	1
	0
	0
	0
	1
	1


Il risultato è che tutta l’espressione è vera solo se a e b sono entrambi falsi (cioè se a < 0 e b non è tra 5 e 7) o se sono entrambi veri (a >= 0 e b è tra 5 e 7).

Questo metodo ci fa capire che quando gli operatori logici sono 2, non importa quanto è complessa l’espressione, i risultati possibili sono sempre 4. Quando sono 3 i risultati sono sempre 8, quando sono 4 i risultati sono sempre 16 e così via.

Purtroppo non è sempre possibile ridurre un’espressione booleana a operatori logici ben definiti come abbiamo fatto sopra, ma quando è possibile e in certi casi anche quando non lo è le tabelle di verità possono costituire un valido aiuto

Cos’è la valutazione short-circuit delle espressioni booleane

Valutando un’espressione booleana il compilatore si blocca non appena può determinare il risultato dell’espressione. Questa caratteristica prende il nome di valutazione a corto-circuito. Per esempio in 

var

  s: string;

...

begin

  ...

  if (Length(s) >= 3) and (s[3] = ‘A’) then ...

se Length(s) < 3 la prima parte dell’espressione è falsa. A questo punto non importa se la seconda parte è vera o falsa, l’espressione nel suo contesto sarà falsa. In tal caso, con la valutazione short-circuit attivata, il compilatore non eseguirà il secondo confronto. Questa peculiarità si rivela molto utile per trucchetti come quello che abbiamo utilizzato nell’esempio appena visto. Poiché leggere un carattere oltre il limite della stringa porterebbe a un errore in fase di esecuzione, effettuando prima il confronto ci accertiamo che il confronto s[3] = ‘A’ venga eseguito solo se s ha almeno 3 caratteri.

La valutazione short-circuit è regolata dalla direttiva di compilatore {$B+/-} o {$BOOLEVAL ON/OFF} che ha ambito d’azione locale ed è attiva di default ({$B+}

Stringa

L’unico operatore stringa è quello di concatenazione.

Per esempio l’istruzione:

S := ‘Ciao’ + ‘ ‘ + ‘mondo’;

riempirà s con la stringa ‘Ciao mondo’. Quando si concatenano stringhe è possibile inserire dei codici di controllo con #<codice ascii> attaccandoli all’inizio o alla fine di un letterale stringa senza nessun spazio vuoto. Per esempio

s := ‘Linea 1’#10’Linea 2’;

concatena e memorizza in s ‘Linea 1’, un carattere di a capo e ‘Linea 2’. La seguente linea:

s := ‘Linea 1’ #10 ’Linea 2’;

provoca invece un errore di compilazione. Per mettere un apice in una stringa mettere i doppi apici

writeln(‘Che ora e’’?’); // scrive “che ora e’?”

writeln(‘’’’); // scrive “’”

Operatori di confronto

Gli operatori di confronto sono:

	Operatore
	Operazione
	Tipi operandi
	Risultato
	Esempio

	=
	Uguale
	simple, class, class reference, interface, string, packed string
	Boolean
	I = Max

	<>
	Diverso
	simple, class, class reference, interface, string, packed string
	Boolean
	X <> Y

	<
	Minore
	simple, string, packed string, PChar
	Boolean
	X < Y

	>
	Maggiore
	simple, string, packed string, PChar
	Boolean
	Len > 0

	<=
	Minore o uguale
	simple, string, packed string, PChar
	Boolean
	Cnt <= I

	>=
	Maggiore o uguale
	simple, string, packed string, PChar
	Boolean
	I >= 1


Come si vede dalla tabella possono essere applicati a tutti i tipi ordinali (interi, carattere, booleani, enumerated, subrange), ai reali e alle stringhe (a diversi tipi di stringa). In realtà gli stessi simboli si utilizzano anche per gli insiemi ma con diversi significati. Non si applicano a record e array. I simboli ‘=’ e ‘<>’ si applicano anche a diversi tipi di puntatore, tra cui i puntatori a classe.

Sui caratteri vale l’ordinamento all’interno del codice ASCII. Per le stringhe viene fatto un vero e proprio confronto di ordine alfabetico utilizzando l’ordine naturale del codice ASCII (per cui in realtà non si tratta di un confronto alfabetico accurato, per esempio le minuscole vengono considerate prima di tutte le maiuscole, cioè ‘z’ < ‘A’). Per i tipi ordinali in genere vale il valore ordinale (per esempio negli enumerated vale la posizione in cui sono definiti i valori).

Precedenze

Le espressioni complesse vengono eseguite applicando gli operatori elementari che le compongono secondo la priorità suggerita dalle regole di priorità e dalle parentesi esplicitamente indicate.

Per esempio in

6 + 7 * 5

la moltiplicazione viene eseguita prima della somma, secondo le normali regole aritmetiche, mentre in

(6 + 7) * 5

viene eseguita prima la somma. Le regole di precedenza degli operatori servono quando mancano le parentesi. A parità di precedenza vale l’associazione da sinistra a destra. Per esempio:

60 / 3 * 2 = 40

poiché viene eseguita prima la divisione, mentre:

60 / (3 * 2) = 10

poiché viene eseguita prima la moltiplicazione. Tra gli operatori aritmetici *,/,div,mod hanno la priorità più alta, +,- la priorità più bassa. Tra gli operatori logici not ha priorità più alta, and media e or/xor bassa. Gli operatori di confronto hanno la priorità più bassa: questo permette di scrivere confronti tra espressioni complesse senza usare le parentesi.

La tabella delle precedenze degli operatori che abbiamo visto finora è la seguente:

	Operatori
	Precedenza

	Not
	1 (più alta)

	*, /, div, mod, and
	2

	+, –, or, xor
	3

	=, <>, <, >, <=, >=
	4 (più bassa)


Si consulti la pagina ‘Operator precedence rules’ della guida in linea per avere la tabella completa

N.B.: si noti che 

if a >= 3 and a <= 6 then...

Darà errore e che si deve scrivere invece:

if (a >= 3) and (a <= 6) then...

Procedure e funzioni

Un’esigenza che si avverte prepotentemente scrivendo codice anche solo di poche decine di righe è quella di avere strumenti che permettano di spezzare un listato monolitico in unità più piccole logicamente complete e indipendenti dal resto del programma, in modo che sia più semplice leggere il codice (che si riduce a una lista di azioni ad alto livello) ed eseguirne la manutenzione. Il processo di dividere un programma in parti più piccole prende il nome di modularizzazione mentre le parti sono chiamate moduli. 

Storicamente la prima soluzione che si è trovata a questo problema è stata l’introduzione delle procedure, o routine o subroutine, nate insieme ai linguaggi ad alto livello. Una procedura è un insieme di istruzioni delimitate in un blocco e identificata da un nome (o identificatore) univoco
. Mentre nei primissimi linguaggi di programmazione tutte le variabili erano globali, con l’affermarsi della programmazione strutturata alla fine degli anni ’60 nacque il concetto di variabile locale, variante della variabile globale il cui ambito d’azione è limitato alla procedura nella quale viene dichiarata.

In Pascal esistono 2 tipi di subroutine: le procedure e le funzioni. Nel corso della dispensa ci riferiremo a entrambe col nome collettivo di routine (ma talora userò anche il termine procedura per riferirmi a entrambe).

La sintassi di una procedura è la seguente:

procedure procedureName(parameterList); directives;

localDeclarations;

begin

  statement1
  ...
  statementN

end;

La sintassi di una funzione è la seguente:

function procedureName(parameterList): returnType; directives;

localDeclarations;

begin

  statement1
  ...
  statementN

end;

Si noti che oltre alla keyword che introduce la definizione (procedure o function) l’unica differenza tra le due è che nella prima riga la funzione ha un tipo di ritorno che manca alla procedura.

La prima riga di una procedura o di una funzione si chiama intestazione (heading o header) o prototipo (prototype) ed esprime in un qualche modo la sua interfaccia. In essa compare il nome, la lista di parametri in input e, nel caso delle funzioni, il tipo del valore di ritorno. Questa informazione è in teoria tutto quello che serve al programmatore per utilizzarla. Tutto quello che segue (le dichiarazioni locali e le istruzioni delimitate dalle parole begin..end) prende il nome di corpo o blocco della routine e ne costituisce l’implementazione.

Prendiamo ad esempio una funzione “Cubo”  che accetta un numero reale come parametro e ritorna quello stesso numero elevato al cubo. Questa funzione ha:

· un nome (Cubo) che il programmatore dovrà utilizzare per richiamarla

· una lista di parametri costituita da un solo parametro di tipo real (chiamiamolo ANum)

· un tipo di ritorno, real

Questa è l’interfaccia della funzione. In più la funzione avrà un’implementazione, ossia del codice che esegue fisicamente i calcoli. Come sia fatto questo codice non è una cosa che interessi al programmatore che deve usare la funzione, che è esclusivamente interessato a cosa fa e a come utilizzarla. Tipicamente la funzione sarà definita in questo modo:

function Cubo(ANum: real): real;

begin

  Cubo := ANum * ANum * ANum

end;

Se si decidesse di cambiare l’implementazione della funzione utilizzando la funzione Power() di Delphi la definizione sarebbe simile a questa:

function Cubo(ANum: real): real;

begin

  Cubo := Power(Anum,3)

end;

Questa modifica non riguarderebbe l’interfaccia della funzione, che resterebbe identica  a prima. Cambierebbe solo il come viene fatto il calcolo ma non cambierebbe né il cosa viene fatto né il come la funzione si interfaccia al programma esterno.

Questa separazione concettuale tra l’interfaccia e il comportamento logico di una funzione da una parte e la sua implementazione dall’altra prende il nome di modello della scatola nera. Riferendoci all’esempio del cubo fatto sopra tutto quello che l’utilizzatore della funzione dovrebbe conoscere è che prende in input un argomento reale e che in output fornisce il numero reale che è il cubo dell’argomento in input. Analogamente tutto quello che si deve sapere di una funzione Somma è che in input prende 2 argomenti di un certo tipo numerico e che in output viene fornita la somma degli argomenti in input. Come vengono fatti i calcoli fisicamente non ha nessuna importanza: le funzioni sono considerate come scatole nere (black box), macchinette dal funzionamento misterioso che prendono dentro dei valori e buttano fuori un risultato

Spesso occorre che una procedura o una funzione ritorni più di un valore. Tra breve vedremo com’è possibile farlo. Prima però vediamo come utilizzare le routine nel programma principale

Chiamata di procedura

Per utilizzare una procedura o una funzione occorre richiamarla da qualche parte nel programma. La chiamata di una procedura viene effettuata col nome seguito dalla lista di parametri effettivi tra parentesi. Per esempio data la procedura

procedure NumString(N: Integer; var S: string);

var

  V: Integer;

begin 

  V := Abs(N);

  S := '';

  repeat

    S := Chr(V mod 10 + Ord('0')) + S;

    V := V div 10;

  until V = 0;

  if N < 0 then S := '-' + S;

end;

La chiamata può essere simile a questa:

NumString(17, MyString);

Nell’esempio sopra N e S sono chiamati parametri formali, mentre 17 e Mystring sono i parametri effettivi (actual parameters) della chiamata. I parametri formali possono essere utilizzati solo all’interno della procedura e nelle sue sottoprocedure e sono sotto molti punti di vista dichiarazioni locali. Il loro scopo è quello di passare l’input alla procedura e, come vedremo tra breve, talora anche di restituire l’ouput.

La chiamata di una funzione è in tutto e per tutto identica a quella di una procedura; in più la chiamata stessa può essere utilizzata all’interno di espressioni dove si identifica con il valore ritornato. Per esempio la funzione Cubo definita sopra può essere utilizzata nei seguenti modi:

var

  X,R: real;

begin

  R := 3.5;

  X := Cubo(R); // chiamata a Cubo

  Writeln(‘Il cubo di 3.5 e’’’,X);

  Writeln(‘Il cubo di 5.1 e’’’,Cubo(5.1)); // chiamata a Cubo

end

Ci sono certi casi in cui al programmatore può non interessare il valore di ritorno della funzione. In tal caso è possibile chiamare la procedura come se fosse una funzione

Dichiarazioni locali e scope di una variabile

In una procedura o funzione tra la prima riga (il prototipo) e le istruzioni della procedura (delimitate tra le parole begin..end) stanno le dichiarazioni locali. In una routine è possibile dichiarare variabili, costanti, tipi di dati, label proprio come nel programma principale. Queste dichiarazioni sono locali alla procedura e valgono solo nell’ambito del codice della procedura e delle sue sottoprocedure.

Proprio come è possibile dichiarare variabili, costanti e tipi di dati è possibile anche dichiarare procedure e funzioni locali che potranno essere richiamate solo dalla procedura stessa, dalle sottoprocedure dello stesso livello dichiarate di seguito e dalle sotto-sottoprocedure di queste ultime. Si ha insomma un albero di procedure, con il programma stesso come radice dell’albero e in cui ciascuna delle procedure appartiene a una procedura genitore e può avere procedure figlie.

Le procedure locali non si usano spesso, specie con la programmazione strutturata. Spesso è più comodo dichiarare sottoprocedure a livello di programma, in modo che siano richiamabili da più procedure.

Lo scope di una varabile è il suo ambito d’azione, ossia il pezzo di codice all’interno del quale può essere usata. Come accennato sopra l’ambito d’azione di una variabile è la procedura all’interno della quale viene dichiarata e tutte le sottoprocedure.

Una variabile dichiarata a livello di programma è detta variabile globale, una variabile dichiarata a livello di procedura è detta variabile locale. Una variabile locale definita con lo stesso nome di una variabile globale o di una variabile locale di ordine superiore nasconde la variabile precedente. E’ comunque sempre possibile riferirsi alle variabili globali qualificandole col nome dell’unità o del programma, ossia premettendo il nome del programma e dell’unità seguito da un puntino.

Esempio:

program Scope;

{$APPTYPE CONSOLE}

uses

  SysUtils;

var

  x: integer;

  procedure LocalProc;

  var

    x: integer;

  begin

    x := 6;

    writeln('Variabile locale x: ',x);

    writeln('Variabile globale x: ',Scope.x);

  end;

begin

  writeln('SCOPE DI UNA VARIABILE');

  writeln;

  x := 5;

  LocalProc;

  readln;

end.

L’output di questo programma è:

SCOPE DI UNA VARIABILE

Variabile locale x: 6

Variabile globale x: 5

Ritorno di funzione e ricorsione

Esistono due sintassi che permettono di ritornare un valore da una funzione. La prima,  che abbiamo visto sopra nel corpo della funzione cubo, consiste nell’assegnare al nome della funzione il valore di ritorno, p.e.:

function Cubo(ANum: real): real;

begin

  Cubo := ANum * ANum * ANum

end;

Una seconda sintassi permette di utilizzare la variabile speciale Result automaticamente definita all’interno di ogni funzione con lo stesso tipo del valore di ritorno

function Cubo(ANum: real): real;

begin

  Result := ANum * ANum * ANum

end;

In entrambi i casi è possibile fare nel corso della funzione più di un assegnamento alle due variabili speciali: il valore ritornato sarà quello dell’ultimo assegnamento eseguito. E’ poi pacifico che modificando il valore di una variabile si modifichi anche il valore dell’altra, poiché entrambe contengono il valore da ritornare. Le due sintassi non sono completamente equivalenti: se si trovano a destra del simbolo di assegnamento il nome della funzione esegue una chiamata ricorsiva alla funzione stessa, mentre Result leggerà il valore della variabile precedentemente assegnato con uno dei due metodi descritti.

La ricorsione, ossia la chiamata di una funzione a sé stessa, è una tecnica avanzata di programmazione che non dovrebbe mai essere utilizzata da programmatori principianti. Viene usata in certe applicazioni matematiche e in Intelligenza Artificiale, per l’esplorazione di grafi e alberi.

A titolo di esempio  riporto qui una funzione che calcola il fattoriale di un numero:

program Fattoriale;

{$APPTYPE CONSOLE}

uses

  SysUtils;

function Fatt(x :integer): integer;

begin

  if x<>1 then result := x * Fatt(x-1)

  else result := 1

end;

var

  n: integer;

begin

  writeln('FATTORIALE');

  writeln;

  write('Inserisci il numero di cui vuoi calcolare il fattoriale: ');

  readln(n);

  writeln('Il fattoriale e'': ', Fatt(n));

  writeln;

  readln;

end.

N.B.: Result non appartiene al Pascal standard. Se volete scrivere codice Pascal portabile utilizzate solo il nome della funzione per ritornare un valore.

Passaggio di parametri

La lista di parametri definisce una lista di identificatori locali di variabili o costanti, la cui sintassi è simile a quella della dichiarazione delle variabili: ogni dichiarazione è separata dalla successiva da un punto e virgola, mentre variabili dello stesso tipo possono essere raggruppate insieme separandole con la virgola. Ecco come potrebbe essere la dichiarazione di una procedura che prende 2 parametri interi, uno stringa e un intero

procedure MyProc(a,b: integer; c: string; d: integer);

...

Esistono quattro tipi di passaggi parametri: per valore, per riferimento, con parametri costanti e con l’utilizzo della parola out.

Il normale passaggio di parametri è chiamato passaggio per valore. Il valore del parametro effettivo (quello nella chiamata) viene copiato nel parametro formale (quello nella dichiarazione). Al termine dell’esecuzione della procedura il valore presente nel parametro formale viene buttato.

Oltre al passaggio di parametri per valore esiste il passaggio per riferimento, ottenibile premettendo all’identificatore del parametro formale la parola riservata var nella lista dei parametri. Col passaggio per riferimento invece di copiare il valore del parametro effettivo in una variabile locale viene passato un puntatore al parametro effettivo stesso. Questo significa che ogni modifica al parametro formale si riflette immediatamente al parametro effettivo; va da sé che il parametro effettivo non può essere un valore costante ma deve essere necessariamente una variabile. Il passaggio per riferimento fornisce così la possibilità di restituire indietro al programma chiamante dei valori diversi da quello ritornato dalla funzione. Il codice che segue esemplifica la differenza tra i due tipi di passaggio parametri, chiamando una PassByRef che prende un argomento per riferimento e una  PassByVal che prende un argomento per variabile.

program ParametriTest;

{$APPTYPE CONSOLE}

uses

  SysUtils;

procedure PassByRef(var x: smallint);

begin

  x := x + 1;

  writeln('Valore in PassByRef: ',x);

end;

procedure PassByVal(x: integer);

begin

  x := x + 1;

  writeln('Valore in PassByVal: ',x);

end;

var

  y: smallint;

begin

  writeln('PASSAGGIO PARAMETRI');

  writeln;

  y := 5;

  writeln('Valore iniziale della variabile globale: ',y);

  PassByVal(y);

  writeln('Valore della variabile globale dopo PassByVal: ',y);

  PassByRef(y);

  writeln('Valore della variabile globale dopo PassByRef: ',y);

  readln;

end.

L’output di questo programma è:

PASSAGGIO PARAMETRI

Valore iniziale della variabile globale: 5

Valore in PassByVal: 6

Valore della variabile globale dopo PassByVal: 5

Valore in PassByRef: 6

Valore della variabile globale dopo PassByRef: 6

Questo esempio dovrebbe chiarire come il passaggio della y a PassByVal lascia inalterata la variabile anche se il valore del parametro corrispondente viene modificato all’interno della procedura, mentre il passaggio della y a PassByRef fa si che i cambiamenti del parametro formale x all’interno della procedura si ripercuotano sul parametro effettivo y. 

I parametri passati per riferimento rompono il concetto di black box: le variabili passate a una procedura per riferimento possono essere modificate in modo inatteso dal chiamante. Vanno perciò utilizzate solo quando è strettamente necessario e il loro comportamento deve essere ben documentato.

Esiste un tipo particolare di passaggio per riferimento che si ottiene usando la parola out invece di var. Se si utilizza out invece di var il comportamento è identico, ma il valore della variabile viene scartato prima di entrare nella funzione. In un buon stile di programmazione è meglio usare out invece di var per sottolineare che l’unica funzione di quel parametro è fornire in output un valore indietro al chiamante mentre non può essere utilizzato per passare un valore alla funzione

Esiste infine un ultimo tipo di passaggio parametri che è quello costante, ottenuto premettendo const al nome della variabile. Il passaggio per costante è un tipo speciale di passaggio per valore in cui il valore del parametro formale non può essere modificato.

Nota

Quando si passa un parametro per riferimento i tipi del parametro formale e di quello effettivo devono coincidere esattamente, mentre nel passaggio per valore devono solo essere compatibili.

Parametri di default

Esiste una sintassi che permette di assegnare valori di default ai parametri di una procedura o di una funzione. Per esempio in:

procedure FillArray(A: array of Integer; Value: Integer = 0);

la procedura FillArray prende due valori: il primo è un array di interi mentre il secondo (Value) è un valore con cui riempire l’array ed ha il valore di default 0.

Il chiamante può non specificare il valore per i parametri che hanno un valore di default, accettando così il valore stesso. Per esempio:

FillArray(A); // riempie A con il valore 0

FillArray(A,5); // riempie A con il valore 5

Il parametro di cui si specifica il valore di default deve essere separato dagli altri (non è possibile specificare un valore di default di una lista di parametri). Per esempio

MyProc(a,b: integer = 5); // Errore!

è sbagliato.

Tutti i parametri con un valore di default devono stare in fondo alla lista. Non è possibile avere dichiarazioni del genere:

MyProc(a: integer; b: char = ‘a’; c: char);

sono sbagliate. La chiamata MyProc(6,,’b’) non è legittima

Dichiarazioni forward

Una procedura non può essere utilizzata prima di essere dichiarata. In altre parole tutte le chiamate alla procedura devono seguire la sua dichiarazione. Se questo è pacifico per le chiamate nel programma principale, questo comporta che tutte le procedure che utilizzano altre procedure devono essere dichiarate dopo di esse. Certe tecniche di programmazione richiedono che due procedure si chiamino l’un l’altra. In questi casi necessariamente una chiamata dovrà precedere la dichiarazione.

Per ovviare a questo inconveniente si fa uso delle dichiarazioni forward. Le dichiarazioni forward sono come definizioni normali di procedura ma hanno la parola riservata forward al posto del corpo della funzione. Una dichiarazione forward specifica perciò un prototipo completo di procedura. A una dichiarazione forward deve seguire una dichiarazione vera più avanti nel codice

Per esempio:

program ForwardTest;

{$APPTYPE CONSOLE}

uses

  SysUtils;

procedure B; forward;

procedure A;

begin

  Writeln('Executing A...');

  B;

end;

procedure B;

begin

  Writeln('Executing B...');

end;

begin

  A;

  readln;

end.

Le dichiarazioni di procedure anticipate da una dichiarazione forward non devono specificare la lista di parametri e se sono funzioni neppure il valore di ritorno, ma se lo fanno le due dichiarazioni devono coincidere esattamente, anche nei nomi dei parametri formali

Dati e tipi di dati

Esistono tipi di dati predefiniti nel linguaggio e tipi di dati costruiti dall’utente. In questa sezione analizziamo i tipi di dati predefiniti.

I tipi di dati predefiniti sono classificabili nelle seguenti categorie:

· semplici

· ordinali

· interi

· carattere

· boolean

· enumerated

· subrange

· reali

· stringa

· strutturati

· insiemi

· array

· record

· file

· altri (classi, riferimenti a classe, interfacce)

· puntatore

· altri (procedurali, variant)

I tipi semplici definiscono insiemi ordinati di valori e sono composti da ordinali e numeri in virgola mobile (real).

Ordinali

I tipi ordinali sono tutti quei tipi in cui ogni elemento ha un successore (tranne l’ultimo) e un predecessore (tranne il primo). I numeri reali sono ordinati ma non sono ordinali, perché sono teoricamente infiniti (in realtà i numeri in virgola mobile sono finiti ma il predecessore e il successore di un numero non sono comunque definiti). Tutti i tipi interi, carattere, boolean, enumerated e subrange sono ordinali.

A un valore ordinale è associato un numero ordinale, che indica la posizione all’interno dell’insieme di valori. Per gli interi il numero ordinale è l’intero stesso, per i caratteri è il suo codice ASCII, per i boolean 0 o 1, per gli enumerated la posizione all’interno della lista partendo da 0, per i subrange il numero ordinale del valore nel tipo di partenza.

Esistono alcune funzioni definite su tutti i tipi ordinali:

	Funzione
	Parametro
	Valore di ritorno
	Note

	Ord
	Espressione ordinale
	Fornisce il numero ordinale
	non prende argomenti Int64

	Pred
	Espressione ordinale
	Valore precedente
	

	Succ
	Espressione ordinale
	Valore successivo
	

	High
	Identificatore di un tipo ordinale o variabile di tipo ordinale
	Valore più alto nel tipo
	Opera anche su stringhe corte e array

	Low
	Identificatore di un tipo ordinale o variabile di tipo ordinale
	Valore più basso nel tipo
	Opera anche su stringhe corte e array


Vediamo queste categorie una ad una

Interi

Esistono diversi tipi di interi a seconda della lunghezza (1, 2, 4 o 8 byte) e del fatto se sono segnati o non segnati.

	
	1 byte
	2 byte
	4 byte
	8 byte

	Segnati
	ShortInt
	SmallInt
	LongInt, Integer 
	Int64

	Non segnati
	Byte
	Word
	Longword, Cardinal
	-


Di solito si usa sempre gli integer se si desidera un numero segnato e un cardinal nel caso in cui si desideri un numero non segnato.

Quando per qualsiasi motivo occorre risparmiare memoria (ad esempio all’interno dei record) si può usare un intero più corto a 1 o a 2 byte. Se si manipola numeri molto grandi può invece essere necessario usare un Int64. In ogni caso un tipo diverso da quello standard può rendersi necessario nell’uso delle API o di funzioni di libreria che richiedano interi ben precisi.

Caratteri

Esistono due tipi carattere: char (o AnsiChar) e WideChar. I char contengono caratteri a 1 byte (ASCII), i WideChar caratteri a 2 byte (Unicode). Nel corso utilizzeremo sempre i char.

Con i caratteri si utilizza la funzione chr che converte un intero in carattere (per esempio chr(65) è ‘A’). Chr è per i caratteri la funzione inversa di Ord (ord(‘A’) è 65)

Booleani

Di solito si utilizza il tipo boolean. Il tipo boolean occupa 1 byte ed ha i valori false (numero ordinale 0) e true (numero ordinale 1).

Per compatibilità con altri linguaggi e sistemi operativi serve a volte usare i seguenti tipi:

	1 byte
	2 byte
	4 byte

	byteBool 
	wordBool
	longBool


Questi tre tipi sono false quando il loro numero ordinale è 0, sono true quando il loro numero ordinale è non 0. L’unica differenza tra byteBool e boolean è questa: boolean è true solo quando il suo numero ordinale è 1, byteBool quando è non zero

Enumerated

Un tipo enumerated (di enumerazione) definisce un insieme di valori semplicemente listando gli identificatori di quei valori. Per esempio:

type

  Seme = (denaro, coppe, spade, bastoni);

var

  SemeCarta: Seme;

Ovviamente un tipo enumerated può essere definito anche all’interno della dichiarazione di una variabile. Per esempio:

var

  SemeCarta: (denaro, coppe, spade, bastoni);

Gli identificatori non possono essere ridefiniti. Così la riga di codice:

var

  SemeCarta1: (denaro, coppe, spade, bastoni);

  SemeCarta2: (denaro, coppe, spade, bastoni);

darà errore. In tal caso si dovranno o definire insieme le due variabili

var

  SemeCarta1, SemeCarta2: (denaro, coppe, spade, bastoni);

o dare una definizione di tipo

Subrange

I subrange sono sottoinsiemi di altri tipi ordinali e vengono definiti con una sintassi del tipo

<primo valore>..<ultimo valore>

Per esempio

type

  giorno = 1..31;

  carattereMinuscolo = ‘a’..’z’;

o anche:

type

  tColors = (Red, Blue, Green, Yellow, Orange, Purple, White, Black);

  TMyColors = Green..White;

Il numero ordinale associato a un valore subrange è identico a quello del tipo originale (nell’esempio sopra il valore ordinale di Green è 2, anche se la variabile è di tipo tMyColors).

Si noti che nessun controllo sui limiti viene fatto a tempo di esecuzione. Questo significa che ogni espressione variabile che debordi dai limiti del subrange si comporterà come un’espressione del tipo originale. Per esempio, mentre le righe di codice:

var

  G: giorno;

begin

  g := 35;

// errore di compilazione!

  g := 31 + 2;
// errore di compilazione!
  …

forniscono un errore di compilazione, il programma:

var

  g1,g2: giorno;

begin

  g1 := 30;



  g2 := 20;

  g1 := g1 + g2;

  writeln(g1);
non fornisce nessun errore, né durante la compilazione né in fase di esecuzione, ma stampa il valore 50, memorizzato nella variabile g1.

Il controllo del debordamento di un tipo subrange a tempo di esecuzione può essere forzato usando la direttiva di compilazione {$R} (tale direttiva abilita i controlli run-time sul range).

Reali

I numeri reali, chiamati anche  numeri in virgola mobile (floating point), sono una famiglia di tipi di dati predefiniti che serve per memorizzare numeri con cifre decimali. Mentre le operazioni su interi vengono eseguite direttamente a livello di linguaggio macchina, le operazioni in virgola mobile sono eseguite da una speciale unità chiamata FPU (Floating Point Unit). La FPU era un coprocessore distinto dalla CPU fino al 386 (il coprocessore matematico), mentre dai 486 e negli attuali Pentium è incorporata all’interno della CPU. Anche se questa evoluzione ha reso i calcoli in virgola mobile molto più rapidi, la matematica reale resta comunque centinaia di volte più lenta di quella intera, per cui va utilizzata solo quando strettamente necessario. Con l’esperienza ci si accorgerà che i numeri in virgola mobile sono molto meno necessari di quanto si possa pensare.

Come per gli interi, ne esistono vari tipi, diversi tra loro per la lunghezza.

	
	4 byte
	6 byte
	8 byte
	10 byte

	Reali
	Single
	Real48
	Real= Double, Currency
	Extended


A differenza degli interi i numeri in virgola mobile sono sempre segnati.

Fino a Delphi 3 il tipo più usato era il Real48, che aveva il nome di real. Esisteva così il single (4 byte), il real (6 byte) e il double (8 byte). Siccome questo formato era proprietario di Delphi e sconosciuto nell’ambito degli altri ambienti di programmazione è stato abbandonato. In Delphi 5 Real denota lo stesso identico tipo di Double, mentre il Real originale è stato ridenominato Real48 e resta per motivi di compatibilità. Se si desidera una maggiore precisione occorre utilizzare Extended, mentre utilizzando Single si avrà meno precisione ma si occuperà meno memoria.

Currency è un secondo formato decimale a 8 byte ed è in realtà un intero a virgola fissa studiato per contenere dati monetari. La parte decimale è capace di mantenere fino a 4 cifre decimali senza errore. Esiste infine un formato a 8 byte non citato nella tabella, il Comp, che è obsoleto e mantenuto solo per scopi di compatibilità all’indietro.

Numeri decimali a virgola mobile e a virgola fissa

Anche se sopra si è lasciato intendere che numeri a virgola mobile e numeri decimali sono la stessa cosa, la realtà è che la virgola mobile è una tecnica precisa utilizzata per memorizzare i numeri decimali. Esistono sostanzialmente due possibili tecniche per memorizzare un numero decimale: usare una virgola fissa (fixed point) o usare una virgola mobile (floating point). 

La virgola fissa è la più semplice. Supponiamo mi servano al massimo 4 cifre decimali. Quello che devo fare è memorizzare il numero moltiplicato per 10.000.

Per esempio:

1 

( 10.000

1,5 

( 15.000

6,7023 
(  67.023

1520,001 
( 15.200.010

Questa è esattamente la tecnica utilizzata dal tipo Currency.

Questa tecnica è poco flessibile, perché di solito occorre memorizzare numeri di cui non si conosce la lunghezza della parte decimale. La tecnica della virgola mobile consiste allora nel memorizzare tutte le cifre significative (chiamate mantissa) più l’esponente a cui deve essere elevata la mantissa per ottenere il numero originale. Per esempio:

	Numero
	Mantissa
	Esponente
	Operazione

	1
	1
	1
	0.1 * 101

	10,435
	10435
	2
	0.10435 * 102

	0,21
	21
	0
	0.21 * 100

	0,0003400075
	3400075
	-3
	0.3400075 * 10-3


Stringhe

Esistono 4 tipi di stringhe in Object Pascal:

· stringhe corte (ShortString) 

· stringhe lunghe (String o AnsiString)

· WideString 

· Stringhe terminate nulle

Stringhe corte

Le stringhe corte sono quelle del Turbo Pascal
. Sono buffer di caratteri allocati staticamente all’avvio del programma e nella dichiarazione occorre specificarne la lunghezza. Per esempio

var

  Nome: string[20];

La lunghezza specificata nella dichiarazione tra parentesi quadre rappresenta la lunghezza massima. La massima lunghezza possibile è 255 byte. Le stringhe corte hanno un byte in più, il byte 0, che registra la lunghezza reale della stringa. Per esempio in:

var

  Nome: string[20];

begin

  nome := ‘Maria’;

Il byte 0, indicizzato come nome[0], contiene il valore 5 (la lunghezza della stringa ‘Maria’). Esiste un tipo speciale, ShortString, che designa le stringhe corte di lunghezza massima (255 byte + 1 byte di lunghezza).

Normalmente si utilizzano le stringhe lunghe perché occupano lo spazio di memoria ottimale, mentre le stringhe corte hanno il peggiore profilo possibile nell’occupazione di memoria

Nota

La documentazione dice che le short string esistono solo per motivi di compatibilità, ma questo non è vero. Le stringhe corte diventano utili in almeno 2 circostanze: con i file di record e con le DLL

Stringhe lunghe

Le stringhe lunghe sono le stringhe più utilizzate in Delphi. Si dichiarano come le stringhe corte omettendo l’indicazione di lunghezza, per esempio:

var

  Nome: string;

A differenza delle corte le stringhe lunghe sono gestite in maniera dinamica. Questo significa che la variabile contiene solo un puntatore alla memoria che contiene il testo e che tutte le volte che si fa un assegnamento la vecchia stringa viene disallocata e viene allocata la nuova in maniera completamente trasparente al programmatore.

L’utilizzo delle stringhe lunghe è molto comodo e dovrebbero essere usate ogni qualvolta sia possibile. Oltre alla comodità sono implementate in modo da occupare solo la quantità di memoria necessaria

WideString

Non c’è molto da dire. Sono stringhe di WideChar ossia di caratteri Unicode. Non le useremo durante il corso

Stringhe terminate nulle

Le stringhe terminate nulle non hanno un byte di lunghezza come le stringhe corte ma un carattere speciale di fine stringa (lo 0). Sono utilizzate da alcuni linguaggi come il C e supportate da Delphi per l’interfacciamento con le API di Windows, scritte in C. Le vedremo via via che le useremo.

Indicizzazione

Tutte le stringhe possono essere indicizzate per accedere ai singoli caratteri. Per esempio stringa[1] fornisce il primo carattere della stringa “stringa”, stringa[2] fornisce il secondo carattere e così via

Tipi di dati strutturati

I tipi di dati strutturati contengono più di un valore. Esistono 3 tipi strutturati: insiemi, array e record.

Insiemi

Gli insiemi contengono gruppi di  valori non ordinati e non ripetuti. La sintassi per dichiarare un insieme è:

set of <baseType>

dove baseType è un tipo ordinale con non più di 256 valori e i cui valori devono avere numeri ordinali compresi tra 0 e 255. Per esempio

type

  TSomeInts = 1..250;

  TIntSet = set of TSomeInts;

  TCharSet: set of 'a'..'z';

Ovviamente invece di dichiarare dei tipi di dati nella sezione type è possibilissimo dichiarare variabili direttamente di un tipo insieme, per esempio

var 

  MySet: set of 'a'..'z';

Operatori su insiemi

Tutte le principali operazioni insiemistiche sono supportate. Ecco la tabella

	Operatore
	Operazione
	Tipi operandi
	Risultato
	Esempio

	+
	Unione
	set
	set
	Set1 + Set2

	–
	Differenza
	set
	set
	S - T

	*
	Intersezione
	set
	set
	S * T

	<=
	Sottoinsieme
	set
	Boolean
	Q <= MySet

	>=
	Sovrainsieme
	set
	Boolean
	S1 >= S2

	=
	Uguaglianza
	set
	Boolean
	S2 = MySet

	<>
	Diseguaglianza
	set
	Boolean
	MySet <> S1

	In
	Appartenenza
	ordinal, set
	Boolean
	A in Set1


Array

Un array è un gruppo di dati indicizzato dello stesso tipo, chiamato tipo base. A differenza degli insiemi gli array sono ordinati e all’interno di un array lo stesso valore può comparire più volte (un elemento è reso univoco dal suo indice, non dal suo valore).

Gli array possono essere statici o dinamici, a una dimensione o a molte dimensioni

Array statici monodimensionali

Un array statico monodimensionale ha la seguente sintassi:

array[indexType] of baseType
Il tipo dell’indice è un qualsiasi tipo ordinale che non ecceda i 2 GB di valori. Tipicamente è un subrange di interi.

Alcuni esempi:

var 

  Array1: array[1..100] of Char;
    // array di 100 caratteri
  Array2: array[boolean] of integer; // array di 2 interi

  Array3: array[char] of boolean;    // array di 256 boolean
  Array4: array[‘a’..’z’] of boolean; // array di 26 boolean
Anche se negli esempi sopra ho dichiarato gli array nelle dichiarazioni delle variabili potevo ovviamente definire un tipo array nella sezione type, per esempio:

Type

  MioArray = array[1..100] of Char; 

var 

  Array1: MioArray;
    // array di 100 caratteri
Per accedere agli elementi dell’array si utilizza il nome dell’array seguito dall’indice dell’elemento tra parentesi quadre, per esempio il seguente codice:

Type

  MioArray = array[1..100] of Char; 

var 

  Array1: MioArray;
    // array di 100 caratteri

Begin

  Array1[1] := ‘c’;

  Array1[2] := ‘i’;

  Array1[3] := ‘a’;

  Array1[4] := ‘o’;

  …

end

riempie i primi 4 elementi dell’array.

Con gli array si possono utilizzare le funzioni Low (fornisce l’indice più basso), High (fornisce l’indice più alto) e Length (fornisce la lunghezza dell’array, ossia il numero di elementi).

Per esempio in:

var MyArray: array[‘c’..’f’] of integer;

Low(MyArray) è ‘c’, High(MyArray) è ‘f’, Length(MyArray) è 4. Si noti che la lunghezza dell’array non indica la sua dimensione in byte. Per avere la dimensione in byte occorre fare sizeof(MyArray) (il risultato è 16).

Array statici multidimensionali

Un array multidimensionale ha molti indici invece che uno solo. Per esempio, per rappresentare dati in forma tabellare occorre un array bidimensionale. La sintassi è:

array[indexType1,…,indexTypeN] of baseType

Alcuni esempi:

type 

  TMatrix = array[1..10, 1..50] of Real;

  TCube = array[0..1, 0..1, 0..1] of Integer;

  TStrangeArray = array[Boolean,1..10,TShoeSize] of Integer;

Per accedere a un singolo elemento dell’array si utilizza il nome dell’array seguito dagli indici tra parentesi quadre ossia:

var

  Tabella: array[1..10, 1..5] of integer; // array di 10 righe x 5 colonne
begin

  Tabella[1,1] := 2; // elemento in alto a sinistra := 2
  Tabella[1,2] := 3; // secondo elemento della prima riga := 3
  Tabella[2,1] := 4; // primo elemento della seconda riga := 4

  …

end

Si noti che un array bidimensionale altro non è che un array di array, per cui le seguenti sintassi sono equivalenti:

var

  Tabella: array[1..10, 1..5] of integer; // array di 10 righe x 5 colonne

  Tabella: array[1..10] of array[1..5] of integer; // 10 righe x 5 colonne
Analogamente per accedere a un elemento di un array bidimensionale esistono due sintassi alternative:

  Tabella[1,1] := 2; // elemento in alto a sinistra := 2

  Tabella[1][1] := 2; // elemento in alto a sinistra := 2
Le funzioni Low, High e Length forniscono rispettivamente il valore più basso, quello più alto e il numero di elementi del primo indice. Per esempio nell’array Tabella visto sopra le 3 funzioni forniscono i valori 1,10,10. Al contrario sizeof fornisce la dimensione in byte dell’intero array. Per accedere al valore più basso o più alto e al numero di elementi del secondo indice bisogna passare alle funzioni Low, High e Length l’array col primo indice specificato, per esempio Low(Tabella[1]), High(Tabella[1]) e Length(Tabella[1]). Per accedere al terzo indice occorre specificare i primi due e così via.

A titolo di esempio riporto un programma completo che stampa a video i primi 50 numeri in 10 righe da 5 colonne e che fa uso di un array bidimensionale.

program ArrayTest;

{$APPTYPE CONSOLE}

uses

  SysUtils;

const

  Righe = 10;

  Colonne = 5;

type

  tabella = array[1..Righe,1..Colonne] of integer;

var

  v: tabella;

  i,j: integer;

begin

  writeln('ARRAY STATICO BIDIMENSIONALE');

  writeln;

  writeln('Questo programma utilizza un array statico bidimensionale');

  writeln('La tabella ha dimensioni ',Length(v),'x',Length(v[1]));

  writeln('Ecco i contenuti');

  for i := Low(v) to High(v) do
  begin

    for j := Low(v[1]) to High(v[1]) do
    begin

      v[i,j] := (i-1)*Colonne + j;

      write(i:2,',',j:2,': ',v[i,j] : 2,'   ');

    end;

    writeln;

  end;

  readln;

end.

Array dinamici monodimensionali

Gli array statici hanno una dimensione fissa e vengono caricati insieme al programma eseguibile. Spesso comunque occorre utilizzare array di cui la dimensione non è nota a priori o che cambi a tempo di esecuzione. Per questo motivo i progettisti di Delphi hanno espanso la sintassi originale del linguaggio creando gli array dinamici. Gli array dinamici si dichiarano con la seguente sintassi:

array of basetype

Per esempio:

var MyFlexibleArray: array of Real;

Non basta dichiarare una variabile di tipo array dinamico per poterla usare, occorre prima allocare fisicamente la memoria. Per far questo si utilizza la funzione SetLength, in questo modo:

var 

  MyFlexibleArray: array of Real;

begin

  // qui non si può utilizzare l’array

  SetLength(MyFlexibleArray,10); // alloca spazio per 10 elementi
  // ora si può utilizzare l’array
  ..

end

Una variabile di tipo array dinamico contiene un puntatore all’array stesso. Delphi si preoccupa di gestire automaticamente tutto ciò che riguarda gli array dinamici. Per esempio non c’è nessun bisogno di disallocare manualmente un array dinamico, come si fa di solito quando si alloca memoria dinamica: Delphi libererà la memoria allocata all’uscita del programma. Inoltre, quando si esegue una nuova SetLength Delphi si preoccupa di allocare il nuovo spazio di memoria, copiare i contenuti dal vecchio al nuovo array e disallocare il vecchio spazio. Se si desidera invece semplicemente disallocare l’array, basta assegnare nil a tutte le variabili che fanno riferimento a quello spazio.

Gli array dinamici sono sempre indicizzati con interi che partono da 0. Perciò è possibile accedere ai singoli elementi dell’array visto nell’esempio sopra usando MyFlexibleArray[0], …, MyFlexibleArray[9]. Stando così le cose la chiamata a Low ritorna sempre 0 mentre la chiamata ad High ritorna sempre Length(..)-1 quando si tratta di array dinamici. La chiamata a sizeof(..) ritorna invece sempre 4, che è la lunghezza del puntatore.

Il seguente programma di esempio stampa i primi n numeri a video:

program ArrayTest;

{ Scrive i primi Len numeri utilizzando

un array dinamico monodimensionale }

{$APPTYPE CONSOLE}

uses

  SysUtils;

type

  vettore = array of integer;

var

  v: vettore;

  i: integer;

  Len: integer;

begin

  writeln('TEST DI ARRAY DINAMICO MONODIMENSIONALE');

  writeln;

  write('Dimmi quanti numeri devo stampare: ');

  readln(Len);

  SetLength(v,Len);

  writeln('La lunghezza dell''array e'' di ',Length(v),' elementi');

  writeln('Contenuti:');

  for i := Low(v) to High(v) do
  begin

    v[i] := i;

    writeln(v[i]);

  end;

  readln;

end.

Come spiegato prima, una nuova chiamata a SetLength ridimensiona l’array, tagliandolo se la nuova dimensione è minore della vecchia o espandendolo se è maggiore. Se si verifica un’espansione i nuovi elementi potrebbero contenere valori casuali
. Per esempio il seguente programma:

program TestSetLength;

{$APPTYPE CONSOLE}

uses

  SysUtils;

type

  vettore = array of integer;

var

  v: vettore;

  i: integer;

begin

  writeln('TEST DI SetLength');

  writeln;

  SetLength(v, 20);

  // riempi 20 elementi

  for i := Low(v) to High(v) do v[i] := i;

  // stampa l'array

  for i := Low(v) to High(v) do write(v[i],' ');

  writeln;

  SetLength(v,10); //taglia a 10 elementi

  // stampa l'array

  for i := Low(v) to High(v) do write(v[i],' ');

  writeln;

  SetLength(v,30);

  // stampa l'array

  for i := Low(v) to High(v) do write(v[i],' ');

  writeln;

  readln;

end.

Produrrà il seguente output:

TEST DI SetLength

0 1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17 18 19

0 1 2 3 4 5 6 7 8 9

0 1 2 3 4 5 6 7 8 9 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0

Array dinamici multidimensionali

Per avere array dinamici multidimensionali occorre utilizzare costrutti con array of ripetuti. Per esempio un array bidimensionale di interi si dichiara nel seguente modo:

var

  tabella: array of array of integer;

E’ possibile accedere agli elementi nel solito modo, ossia con tabella[i,j] o tabella[i][j].

Le funzioni Low, High e Length si comportano esattamente come si comportano con gli array statici multidimensionali, mentre sizeof ritorna sempre 4.

La chiamata a SetLength avrà come argomenti l’array più un argomento per ogni dimensione. A titolo di esempio il seguente listato stampa una tabella bidimensionale dei primi n numeri chiedendo all’utente le due dimensioni della tabella:

program ArrayTest;

{$APPTYPE CONSOLE}

uses

  SysUtils;

const

  MaxRighe = 10;

  MaxColonne = 10;

type

  tabella = array of array of integer;

var

  v: tabella;

  i,j: integer;

  righe,colonne: integer;

begin

  writeln('ARRAY DINAMICO BIDIMENSIONALE');

  writeln;

  writeln('Questo programma utilizza un array dinamico bidimensionale');

  repeat

    write('Dammi il numero di righe: ');

    readln(righe);

    if righe > MaxRighe then writeln('Max ',MaxRighe,' righe!');

  until righe <= MaxRighe;

  repeat

    write('Dammi il numero di colonne: ');

    readln(colonne);

    if colonne > MaxColonne then writeln('Max ',MaxColonne,' colonne!');

  until colonne <= MaxColonne;

  SetLength(v,Righe,Colonne);

  writeln('La tabella ha dimensioni ',Length(v),'x',Length(v[1]));

  writeln('Ecco i contenuti');

  for i := Low(v) to High(v) do
  begin

    for j := Low(v[1]) to High(v[1]) do
    begin

      v[i,j] := i*Colonne + j;

      write(v[i,j]:3,' ');

    end;

    writeln;

  end;

  readln;

end.

Si noti la chiamata a SetLength.

Gli array dinamici multidimensionali non devono per forza di cose essere rettangolari. E’ per esempio possibile scrivere un programma che inizializzi un array triangolare di 10 righe, dove la riga 0 ha 0 colonne, la riga 1 ha 1 colonna, la riga 2 ha 2 colonne ecc. fino alla riga 9 che ha 9 colonne. Per fare una cosa del genere occorre dichiarare un array bidimensionale dinamico e usare la SetLength con solo l’argomento relativo al primo indice (SetLength(A,10)) quindi chiamare la SetLength su ogni singola riga dell’array (SetLength(A[0],0), SetLength(A[1],1), …, SetLength(A[9],9). Ecco il listato:

var

  A : array of array of string;

  I, J : Integer;

begin

  SetLength(A, 10);

  for I := Low(A) to High(A) do
  begin

    SetLength(A[I], I);

    for J := Low(A[I]) to High(A[I]) do
      A[I,J] := IntToStr(I) + ',' + IntToStr(J) + ' ';

  end;

end;

Copia e confronto di array

Le operazioni di copia e di confronto hanno significati diversi a seconda che vengano effettuati su array statici o su array dinamici.

La copia di array statici esegue una copia elemento per elemento dall’array a destra del segno di assegnamento all’array a sinistra del simbolo. Per esempio alla fine dell’esecuzione del seguente codice:

var

  A, B: array[1..5] of Integer;

  I: integer;

begin

  for I := 1 to 5 do A[i] := i;

  B := A;

  B[1] := 6;

end;

l’array A contiene i valori [ 1, 2, 3, 4, 5] mentre l’array B contiene i valori [ 6, 2, 3, 4, 5 ]. All’atto dell’assegnazione tutti gli elementi di A vengono copiati in quelli di B, ma le due variabili continuano a puntare a spazi di memoria differente.

Per inciso, due array statici sono compatibili per assegnamento solo se appartengono allo stesso identico tipo. Così il seguente codice:

var

  Int1: array[1..10] of Integer;

  Int2: array[1..10] of Integer;

  ...

Int1 := Int2;

non funzionerà. Per farlo funzionare occorre che Int1 e Int2 siano dichiarate dello stesso identico tipo, in uno dei seguenti due modi:

var Int1, Int2: array[1..10] of Integer;

oppure

type IntArray = array[1..10] of Integer;

var

  Int1: IntArray;

  Int2: IntArray;

La copia di array dinamici è una copia di puntatori. Alla fine del seguente codice:

var

  A, B: array of Integer;

  I: integer;

begin

  SetLength(A, 5);

  for I := 1 to 5 do A[i] := i;

  B := A;

  B[0] := 6;

end;

sia l’array A che l’array B contengono i valori [ 6, 1, 2, 3, 4 ]. Dopo l’assegnamento qualsiasi modifica ad A si riflette su B e qualsiasi modifica a B si riflette su A. Questo succede perché A e B sono diventati riferimenti allo stesso spazio di memoria. Se invece uno dei due viene ridimensionato con SetLength, la riallocazione fa si che le due variabili puntino a due spazi di memoria differente.

Il confronto tra array statici non è possibile. Per gli array dinamici invece viene eseguito non un confronto elemento per elemento ma sui puntatori: se puntano allo stesso oggetto sono uguali, altrimenti sono diversi anche se contengono gli stessi valori

Inizializzazioni

Gli array possono essere inizializzati racchiudendo i valori tra parentesi tonde. Per esempio:

type 

 TCube = array[0..1, 0..1, 0..1] of Integer;

const 

 Digits: array[0..9] of Char = ('0', '1', '2', '3', '4', '5', '6', '7', '8', '9');

 Maze: TCube = (((0, 1), (2, 3)), ((4, 5), (6,7)));

La sintassi delle parentesi tonde è stata probabilmente scelta dai progettisti del Pascal per distinguere le inizializzazioni di variabili insieme da quelle di variabili array. Si veda il paragrafo sulle inizializzazioni per maggiori chiarimenti.

Passaggio di parametri classico

Gli array statici possono essere passati a una procedura o a una funzione dichiarando un tipo array a cui riferirsi nella dichiarazione del parametro formale. Non è possibile dichiarare una procedura in un modo del genere:

procedure Sort(A: array[1..10] of Integer);   // syntax error
ma bisogna per forza dichiarare un tipo array, per esempio:

type TDigits = array[1..10] of Integer;

procedure Sort(A: TDigits);

Nota

La stessa limitazione la abbiamo per tutti i tipi strutturati (record e insiemi)

Open Array e open array constructors

La sintassi classica del passaggio di parametri è un tantino rigida, in quanto obbliga il programmatore a scrivere routine che accettano array di un solo e ben preciso tipo di array. Per rimediare a questo i progettisti Delphi si sono inventati una nuova sintassi, quella dei parametri aperti (open array). La sintassi degli open array si ottiene specificando come tipo di un parametro formale un array of basetype, per esempio:

procedure Sort(A: array of Integer);

Una dichiarazione del genere permette di passare alla funzione variabili di qualsiasi tipo di array monodimensionali di interi, sia statici che dinamici, di qualsiasi lunghezza.

Gli open array hanno sempre indici interi che partono da 0, come gli array dinamici. Ma attenzione a non confondere le due cose: anche gli array statici possono essere passati come parametri aperti. Comunque gli open array rappresentano l’unico modo con cui gli array dinamici possono essere passati a procedure e funzioni.

Gli open array sottostanno alle seguenti regole:

· hanno sempre indici interi che partono da zero, anche quando vengono passati array statici con indicizzazioni particolari. In tal caso l’utilizzo di Low e High sullo stesso array all’interno della procedura o nel programma principale produrrà effetti differenti

· non è possibile il confronto e la copia all’interno della procedura. Questo anche perché alla stessa procedura potrebbero essere passati sia array statici che array dinamici e come abbiamo visto copia e confronto hanno significati differenti nei due casi

· non è possibile utilizzare SetLength (il ché è ovvio visto che SetLength non può essere chiamata su array statici)

Oltre alla sintassi degli open array i progettisti Delphi si sono inventati una sintassi che permette di passare degli array costanti costruiti direttamente nella chiamata e che consiste nello specificare gli elementi racchiusi tra parentesi quadre. Per esempio:

var I, J: Integer;

procedure Add(A: array of Integer);

begin

  …

  Add([5, 7, I, I + J]);

Questi array costanti, chiamati costruttori di open array, possono essere utilizzati come parametri effettivi di open array.

Note

1) Si noti che la sintassi degli open array è adatta solo ad array monodimensionali. Non esiste nessun equivalente per array multidimensionali. Non è possibile per esempio specificare un array bidimensionale di interi in questo modo:

procedure Add(A: array of array of Integer); // errore di compilazione!

2) Le due sintassi: 

type TDigits = array of Integer;

procedure Sort(A: TDigits);

e

procedure Sort(A: array of Integer);

non sono affatto equivalenti. La prima definisce una funzione Sort che accetta solo array dinamici, la seconda definisce una funzione con un parametro aperto, che può cioè prendere sia array statici che dinamici

Record

Un record (chiamato in altri linguaggi struct o structure o data structure) è un tipo di dati composto da elementi eterogenei più semplici. Ogni elemento è chiamato campo (field) ed ha un nome e un tipo di dati. La dichiarazione di un record si riduce alla lista dei suoi elementi. Ecco la sintassi:

type 

  nome = record
    fieldList1: type1;

    ...

    fieldListn: typen;

  end

dove il punto e virgola dopo l’ultimo campo è opzionale. Per esempio:

type

  Data = record
    Anno: Integer;

    Mese: (Jan, Feb, Mar, Apr, May, Jun, 

            Jul, Aug, Sep, Oct, Nov, Dec);

    Giorno: 1..31;

  end;

La dichiarazione del tipo di dati in sé non crea spazio ma occorre dichiarare delle variabili di tipo record, nel solito modo:

var Record1, Record2: Data;

Ovviamente è sempre possibile definire la struttura del record direttamente all’interno della dichiarazione delle variabili, per esempio:

var S: record

  Name: string;

  Age: Integer;

end;

Per accedere ai campi di un record si utilizza la sintassi variabileRecord.Campo. Quando si specifica il campo di un record in questo modo si dice che si qualifica il campo col nome del record. Per esempio il seguente programma stampa a video la data della missione Apollo 11 che portò l’uomo sulla Luna:

program Sbarco;

{$APPTYPE CONSOLE}

uses

  SysUtils;
type

  Data = record
    Anno: Integer;

    Mese: (Jan, Feb, Mar, Apr, May, Jun, 

            Jul, Aug, Sep, Oct, Nov, Dec);

    Giorno: 1..31;

  end;

var

  SbarcoSullaLuna: Data;

begin

  SbarcoSullaLuna.Anno := 1969;

  SbarcoSullaLuna.Mese := Jul;

  SbarcoSullaLuna.Giorno := 20;

  write(‘L’’uomo e’’ sbarcato sulla Luna il ‘);

  writeln(SbarcoSullaLuna.Giorno,’/’,ord(SbarcoSullaLuna.Mese)+1,’/’, SbarcoSullaLuna.Anno);

  readln;

end.

Costrutto with

Poiché ripetere continuamente il nome della variabile può essere scomodo, servirebbe una sintassi che permetta di modificare tutti i campi specificando una sola volta il nome del record a cui ci si riferisce. Questo è esattamente quello che fa il costrutto with. La sintassi di with è:

with record1,…,recordN do statement;

Per esempio:

with SbarcoSullaLuna do 

begin

  Anno := 1969;

  Mese := Jul;

  Giorno := 20;

end;

All’interno dell’istruzione with può esserci qualsiasi istruzione. L’unica differenza delle istruzioni interne al with è che possono riferirsi ai campi del record senza qualificarli ma trattandoli come normalissime variabili. Nell’esempio sopra avremmo per esempio potuto inserire le due istruzioni write all’interno del costrutto, in questo modo:

with SbarcoSullaLuna do 

begin

  Anno := 1969;

  Mese := Jul;

  Giorno := 20;

  write(‘L’’uomo e’’ sbarcato sulla Luna il ‘);

  writeln(Giorno,’/’,ord(Mese)+1,’/’, Anno);

end;

Se si guarda la sintassi si noterà che con with si possono specificare più record insieme. Il seguente codice:

With Record1, Record2, … RecordN do statement
È equivalente a:

With Record1 do
  With Record2 do
    …

      With RecordN do
        statement

In tal caso i campi di Record1,…,RecordN  non hanno bisogno di essere qualificati in statement. Inoltre, in caso di conflitto di nomi, hanno priorità maggiore i campi del with più interno (o equivalentemente quelli del record più a destra all’interno dello stesso with). È perciò assurdo usare il costrutto with per due record dello stesso tipo poiché i campi del record più a sinistra dovranno comunque essere qualificati.

Copia e confronto

I record dello stesso tipo possono essere copiati tramite l’istruzione di assegnamento, per esempio:

var

  A,B: data;

begin

  …

  B := A;

  …

non è invece possibile effettuare un confronto tra record, neppure se sono record dello stesso tipo. Se si desidera confrontare i contenuti di due record occorre farlo elemento per elemento.

Passaggio di parametri

Passare un record a una funzione è semplice. Ecco un esempio:

type

  Tcoord = record
    X,Y: integer;

  end;

procedure VaiAPunto(pos: tcoord);

begin

  …

Se bisogna passare un record a una procedura bisogna necessariamente definire un tipo e specificare il nome del tipo di record come tipo del parametro formale. Una sintassi del genere:

procedure VaiAPunto(pos: record x,y: integer end);

non è ammessa per ovvie ragioni (si pensi a un record che definisce decine di campi).

Si ricordi poi di passare un record per riferimento se si desidera modificarne i campi. Se si passa un record per valore tutte le modifiche fatte ai campi all’interno della procedura andranno perdute.

Inizializzazione

E’ possibile inizializzare un record che sia una variabile globale o una costante tipata facendo seguire alla dichiarazione i valori tra parentesi tonde nel seguente modo:

type

  Data = record
    Anno: Integer;

    Mese: (Jan, Feb, Mar, Apr, May, Jun, 

            Jul, Aug, Sep, Oct, Nov, Dec);

    Giorno: 1..31;

  end;

var d: Data = (Anno: 1996; Mese: Jul; Giorno: 20);

Ogni valore è specificato premettendo il nome del campo seguito dal simbolo ‘:’. Alcuni campi possono essere omessi, ma l’ordine dei campi non può essere invertito. Se si omette di specificare alcuni campi questi saranno inizializzati nella maniera solita (ossia ponendoli a 0)

Parti varianti nei record

Certe volte occorre gestire database di record che sono in gran parte simili tra loro eccetto che per alcuni campi che possono variare. Supponiamo per esempio di avere bisogno di un tipo di record TEmployee che memorizzi per tutti gli impiegati nome, cognome e data di nascita e che in più memorizzi il salario annuale dei dipendenti o la tariffa oraria dei collaboratori pagati a ore. Una possibilità è quella di creare un record che abbia entrambi i tipi di informazione più un boolean che dice se l’impiegato è stipendiato o meno. Per esempio:

type

  TEmployee = record
  FirstName, LastName: string[40];

  BirthDate: TDate;

  Salaried: Boolean

  AnnualSalary: Currency;

  HourlyWage: Currency;

end;

Questa soluzione ha lo svantaggio di sprecare memoria inutile. Infatti o un impiegato è stipendiato e serve il campo AnnualSalary o ha un salario orario e serve il campo HourlyWage ma mai entrambe le cose contemporaneamente. La soluzione ideale sarebbe poter utilizzare uno spazio all’interno del record in maniera flessibile per memorizzare un’informazione o l’altra. Questa soluzione esiste ed è data dalle parti varianti nei record. Un record con una parte variante ha la seguente sintassi:

recordTypeName = record
  fieldList1: type1;

  ...

  fieldListn: typen;

case [tag:] ordinalType of
  constantList1: (variant1);

  ...

  constantListn: (variantn);

end;

La prima parte della definizione del record è quella classica. Il tag se esiste viene gestito come un normalissimo campo della struttura di tipo ordinalType (che è un tipo ordinale). A ogni caso del costrutto case è associato un modo diverso di gestire lo spazio nel record; ogni variant è costituito da una lista di campi con relativi tipi e circondato da parentesi tonde.

Di seguito mostro come modificare il record visto sopra per ottenere una soluzione ottimale grazie a questa sintassi:

type

  TEmployee = record
  FirstName, LastName: string[40];

  BirthDate: TDate;

  case Salaried: Boolean of
    True: (AnnualSalary: Currency);

    False: (HourlyWage: Currency);

end;

Questo record ha i campi fissi FirstName, LastName, BirthDate e Salaried. In più gestisce nello stesso spazio i campi AnnualSalary e HourlyWage:. Il fatto che questi due campi occupino lo stesso spazio significa che modificando uno si modifica l’altro e viceversa.

Il campo Salaried può essere utilizzato come un normalissimo campo fisso del record. Il nome del campo ordinale può essere omesso: in tal caso non verrà creato un campo aggiuntivo.

La parte variabile del record occupa come la variante più ingombrante.

Ecco altri esempi:

type

  TPerson = record
  FirstName, LastName: string[40];

  BirthDate: TDate;

  case Citizen: Boolean of
    True: (Birthplace: string[40]);

    False: (Country: string[20];

            EntryPort: string[20];

            EntryDate, ExitDate: TDate);

  end;

  TShapeList = (Rectangle, Triangle, Circle, Ellipse, Other);

  TFigure = record
    case TShapeList of
      Rectangle: (Height, Width: Real);

      Triangle: (Side1, Side2, Angle: Real);

      Circle: (Radius: Real);

      Ellipse, Other: ();

  end;

Costanti tipate

Le costanti sono normalmente dichiarate senza specificare il tipo. Il compilatore capisce di che tipo è la costante guardando il valore. Per esempio:

Const
  Massimo = 200; // variabile di tipo byte

  StringaPredefinita = ‘vuoto’; // variabile di tipo string

  Pi = 3.14; // variabile di tipo real

Se in una dichiarazione di costante si specifica il tipo si realizza una costante tipata. Le costanti tipate non sono vere costanti, in quanto il loro valore può essere modificato con assegnamenti durante l’esecuzione del programma.

Le costanti tipate sono utili per due ragioni. 

La prima, indicata anche dalla manualistica, è che permettono di inizializzare array e record. Se ad esempio serve un array costante di stringhe è possibile dichiararlo con un’istruzione del genere:

Const

  GiorniDellaSettimana: array[1..7] of string = 

    ‘lun’,’mar’,’mer’,’gio’,’ven’,’sab’,’dom’;

Anche se lo stesso risultato si può ottenere inizializzando una variabile globale, utilizzare una costante tipata rende immediatamente chiaro che il suo valore non cambierà durante l’esecuzione del codice e permette di inserire l’array costante in una dichiarazione locale, cosa non vera per le variabili. È poi possibile rendere una costante tipata una vera costante utilizzando la direttiva {J-} (di ambito locale), che rende le costanti tipate read-only.

Un secondo motivo per cui sono importanti le costanti tipate è che sono l’unico strumento a disposizione in Object Pascal per utilizzare variabili statiche. Le variabili statiche sono variabili locali che si comportano come variabili globali: anche se è possibile utilizzarle solo nell’ambito della procedura in cui sono definite, il loro ciclo di vita è quello del programma, ossia vengono create e distrutte insieme al programma. Così il valore di una variabile statica si conserva di chiamata in chiamata proprio come quello di una variabile globale, senza introdurre gli sgradevoli effetti collaterali che si verificano quando si utilizzano variabili globali all’interno delle procedure. Invece di utilizzare variabili globali all’interno di procedure, è sempre bene prima verificare che non sia sufficiente utilizzare una variabile statica.

Inizializzazioni

Tutte le variabili globali (quelle di programma) e le costanti tipate vengono inizializzate da Delphi a 0. Le variabili locali (locali alle procedure e alle funzioni) contengono invece valori casuali. 

Il programmatore può impostare il valore predefinito delle variabili globali e delle costanti tipate facendo seguire nelle dichiarazioni il valore al tipo in questo modo:

var identifier: type = constantExpression;

Per esempio:

var I: Integer = 7;

Non è possibile inizializzare variabili dichiarate in dichiarazioni multiple (per esempio var i,j: integer = 3) e  non è possibile inizializzare variabili di tipo file e variant.

Type casting

In inglese type cast significa conversione di tipo. A volte è necessario considerare il contenuto di variabili di un certo tipo come un valore di un tipo differente. È possibile fare questo con la conversione di tipo, che ha la seguente sintassi:

typeIdentifier(expression)

Per esempio:

Integer('A')  // ritorna il codice Ascii di ‘A’ (65)
Char(48)   
 // ritorna il carattere con codice Ascii 48 (‘0’)
Boolean(0)
 // ritorna false

Color(2)      // se Color è un enumerated ritorna il terzo valore di Color
Longint(@Buffer) // ritorna l’indirizzo della variabile Buffer come intero

La conversione di tipo non prevede la generazione di codice ma è semplicemente un modo per aggirare il forte controllo dei tipi del Pascal.

E’ possibile fare type casting anche su una variabile, con lo stesso significato. Per esempio:

Char(I)

Boolean(Count)

TSomeDefinedType(MyVariable)

L’operazione di type casting può stare anche a sinistra del simbolo di assegnamento, purché non vengano eseguite conversioni implicite; per esempio:

Var

  I: shortInt;

…

Char(I) := ‘B’; // assegna alla variabile I il codice Ascii di ‘B’ (66)

Un possibile uso di type casting è quello di accedere a una zona di memoria in maniera differente da quella prevista dal tipo di dato. Per esempio:

type

  TByteRec = record
    Lo, Hi: Byte;

  end;

var

  W: Word;

begin

  W := 1234;

  TByteRec(W).Hi := 0;

end;

Questo codice accede al byte più significativo di W senza modificare il byte meno significativo convertendo W al tipo tByteRec. Esistono due funzioni di sistema (Lo e Hi) che accedono al byte più significativo e a quello meno significativo, ma le funzioni non possono essere usate a sinistra del simbolo di assegnamento.

Array e record impacchettati

Consideriamo la seguente struttura dati:

type

  TImpiegato = record
    Nome: string[20];

    Cognome: string[40];

    Eta: integer;

  end;

In teoria ogni variabile di tipo tImpiegato deve occupare 21 + 41 + 4 = 66 byte (si ricordi che le stringhe corte hanno 1 byte in più di lunghezza). Se invece si esegue l’istruzione:

Writeln(sizeof(Timpiegato));

si scoprirà che la dimensione di un record di tipo tImpiegato è 68 byte. Come mai? Il compilatore inserisce all’interno dei record alcuni byte vuoti per ottimizzare il tempo di accesso ai dati. Per esempio i campi di tipo intero vengono allineati ai 4 byte poiché esistono istruzioni in L/M capaci di spostare 4 byte rapidamente solo se risiedono in indirizzi divisibili per 4. Quindi i record di tipo TImpiegato hanno 2 byte di allineamento tra Cognome e Eta affinché il campo Eta inizi al byte 64 invece che al byte 62.

Per dire al compilatore di ottimizzare il consumo di memoria si utilizza la parola chiave packed. Per esempio:

type

  TImpiegato = packed record
    Nome: string[20];

    Cognome: string[40];

    Eta: integer;

  end;

Eseguendo writeln(sizeof(Timpiegato)) si scoprirà che il tipo Timpiegato richiede 66 byte. L’utilizzo della parola riservata packed sui record farà si che Delphi non inserisca mai byte di allineamento. Tale parola può essere usata anche sugli array con lo stesso scopo. Per esempio:

type

  tMyArray = packed array[1..10] of integer;

Comunque con Delphi5 packed non avrà alcun effetto reale sugli array. 

Puntatori

Una variabile di tipo puntatore è una variabile che punta a una zona di memoria con un tipo di dato particolare. Si dichiara premettendo un accento circonflesso al tipo a cui punta, per esempio:

type

  TImpiegato = record
    Nome: string[30];

    Cognome: string[40];

    Eta: integer;

  end;

var

  PImp: ^TImpiegato; // pImp è di tipo puntatore a Timpiegato

Una variabile di tipo puntatore occupa 4 byte e il suo contenuto è un indirizzo a un’area di memoria. Per farla puntare a una variabile statica occorre usare l’operatore @, che fornisce l’indirizzo in cui è allocata la variabile. Per accedere alla variabile puntata dal puntatore si fa seguire il simbolo ^ al nome della variabile puntatore. 

Per esempio il seguente programma:

program Puntatori;

{$APPTYPE CONSOLE}

uses

  SysUtils;

type

  TImpiegato = record
    Nome: string[30];

    Cognome: string[40];

    Eta: integer;

  end;

var

  Imp1: TImpiegato = (Nome: ‘Mario’; Cognome: ‘Rossi’; Eta: 40);

  Imp2: TImpiegato = (Nome: ‘Giorgio’; Cognome: ‘Bianchi’; Eta: 30);

  PImp: ^TImpiegato;

begin

  writeln('PUNTATORI');

  writeln;

  { Utilizzo statico }

  writeln('Utilizzo statico');

  PImp := @Imp1;

  writeln('PImp punta a Imp1: ',PImp^.Nome,' ',PImp^.Cognome,', ',PImp^.Eta);

  PImp := @Imp2;

  writeln('PImp punta a Imp2: ',PImp^.Nome,' ',PImp^.Cognome,', ',PImp^.Eta);

  writeln;

  readln;

end.

Ha come output:

PUNTATORI

Utilizzo statico

PImp punta a Imp1: Mario Rossi, 40

PImp punta a Imp2: Giorgio Bianchi, 30

Si noti l’utilizzo di Pimp^ per accedere ai campi della variabile puntata.

Gestione dinamica della memoria

L’utilizzo della variabile puntatore vista sopra costituisce un metodo di puntamento ai dati alternativo all’indicizzazione. Anche se a prima vista l’indicizzazione può sembrare più potente e flessibile (definisco un array[..] of TImpiegato e navigo tra i record semplicemente incrementando e decrementando un indice intero) è vero l’esatto contrario. Per capire come mai la programmazione con puntatori sia la più flessibile e potente occorre vedere i metodi dinamici di allocazione della memoria.

La memoria può essere allocata in maniera statica o in maniera dinamica. Un esempio di allocazione statica è la dichiarazione di una variabile globale. Una variabile globale viene creata e distrutta insieme al programma: il suo ciclo di vita è il medesimo del programma. Dichiarare variabili globali e costanti tipate è l’unico modo di allocare staticamente della memoria.

L’allocazione dinamica, per contro, viene eseguita manualmente dal programmatore ogni volta che il programma necessita di più memoria. Per esempio si pensi al problema di gestire un database in memoria. Una gestione statica potrebbe prevedere l’allocazione di un array di record. Dichiarando l’array bisogna specificare subito la dimensione massima del database, per esempio:

Var

  DB: array[1..10000] of TImpiegato;

Questo metodo risulta largamente inefficiente. Infatti, ogni volta che il programma viene caricato, si occupa da subito lo spazio per 10.000 record, anche se il database è vuoto. Se per esempio ogni record occupa 68 byte verranno riservati 680 kB solo per il database. Sebbene questo limite appaia alto, potrebbe ancora diventare stretto a chi ha l’esigenza di gestire grossi quantitativi di dati.

L’ideale sarebbe poter usare la memoria strettamente necessaria per i record presenti nel database, non sprecando neppure un byte per memorizzare record vuoti. La gestione dinamica della memoria consente di fare esattamente questo.

Di seguito fornirò l’esempio di utilizzo di 3 tecniche di allocazione dinamica della memoria:

· tramite l’utilizzo di GetMem/FreeMem

· tramite l’utilizzo di AllocMem/FreeMem

· tramite l’utilizzo di New/Dispose

Vediamo subito un esempio pratico che utilizza tutte e 3 le tecniche. Espando il codice del programma visto prima:

program Puntatori2;

{$APPTYPE CONSOLE}

uses

  SysUtils;

type

  TImpiegato = record
    Nome: string[30];

    Cognome: string[40];

    Eta: integer;

  end;

var

  Imp1: TImpiegato = (Nome: ‘Mario’; Cognome: ‘Rossi’; Eta: 40);

  Imp2: TImpiegato = (Nome: ‘Giorgio’; Cognome: ‘Bianchi’; Eta: 30);

  PImp: ^TImpiegato;

begin

  writeln('PUNTATORI');

  writeln;

  { Utilizzo statico }

  writeln('Utilizzo statico');

  PImp := @Imp1;

  writeln('PImp punta a Imp1: ',PImp^.Nome,' ',PImp^.Cognome,', ',PImp^.Eta);

  PImp := @Imp2;

  writeln('PImp punta a Imp2: ',PImp^.Nome,' ',PImp^.Cognome,', ',PImp^.Eta);

  writeln;

  { Utilizzo dinamico con GetMem/FreeMem }

  writeln('Uso di GetMem/FreeMem');

  GetMem(PImp,sizeof(TImpiegato));

  with PImp^ do
  begin

    Nome := 'Pippo';

    Cognome := 'Poppo';

    Eta := 20;

  end;

  writeln(PImp^.Nome,' ',PImp^.Cognome,', ',PImp^.Eta);

  FreeMem(PImp);

  writeln;

  { Utilizzo dinamico con GetMem/FreeMem }

  writeln('Uso di AllocMem/FreeMem');

  PImp := AllocMem(sizeof(TImpiegato));

  with PImp^ do
  begin

    Nome := 'Uto';

    Cognome := 'Ughi';

    Eta := 18;

  end;

  writeln(PImp^.Nome,' ',PImp^.Cognome,', ',PImp^.Eta);

  FreeMem(PImp);

  writeln;

  { Utilizzo dinamico con New/Dispose }

  writeln('Uso di New/Dispose');

  New(PImp);

  with PImp^ do
  begin

    Nome := 'Gustavo';

    Cognome := 'Branduardi';

    Eta := 25;

  end;

  writeln(PImp^.Nome,' ',PImp^.Cognome,', ',PImp^.Eta);

  Dispose(PImp);

  writeln;

  readln;

end.

Il risultato di questo programma è:

PUNTATORI

Utilizzo statico

PImp punta a Imp1: Mario Rossi, 40

PImp punta a Imp2: Giorgio Bianchi, 30

Uso di GetMem/FreeMem

Pippo Poppo, 20

Uso di AllocMem/FreeMem

Uto Ughi, 18

Uso di New/Dispose

Gustavo Branduardi, 25

L’utilizzo di AllocMem e GetMem è molto simile: la differenza, oltre al modo in cui viene ritornato l’indirizzo dello spazio allocato, consiste nel fatto che AllocMem azzera lo spazio di memoria allocato prima di ritornare al chiamante. AllocMem fa praticamente 2 cose: azzera lo spazio di memoria e chiama GetMem. Lo spazio allocato da entrambe le routine viene liberato con FreeMem.

New e Dispose vengono richiamate per allocare una variabile dinamica di un tipo preciso (come nell’esempio sopra). Mentre per richiamare AllocMem e GetMem occorre specificare la dimensione del blocco di memoria che si desidera allocare, con New non è necessario perché New guarda il tipo di dato a cui punta la variabile puntatore. La memoria richiamata con New va liberata necessariamente con un Dispose.

File

In Pascal esistono sostanzialmente 3 tipi di file:

· file di testo

· file tipati

· file non tipati

File di testo

Le variabili di file di testo si dichiarano in uno dei due seguenti modi:

Var f: Text; // modo classico Pascal
oppure

Var f: TextFile; // estensione Borland
Un file di testo è un file composto da righe di testo che può essere aperto solo in lettura o solo in scrittura. Nelle applicazioni console esistono due file di testo dichiarati implicitamente, Input e Output, legati rispettivamente all’input e all’output da e verso console.

Prima di utilizzare una variabile di tipo file occorre associarla a un file esterno
. Per farlo occorre usare la procedura AssignFile, che prende due parametri: la variabile di tipo file e una stringa che designa il percorso e il nome del file esterno. Per esempio:

AssignFile(f,’C:\Prova.txt’);

Il nome del file può essere qualificato dal percorso, come nell’esempio sopra, o senza specificazione di percorso. Se il percorso non è specificato o se è un percorso relativo la directory di riferimento è quella corrente.

La procedura AssignFile non apre il file né ne controlla l’esistenza, ma si limita semplicemente a preparare la variabile perché possa essere usata. Per aprire un file di testo occorre utilizzare una delle seguenti procedure:

· Reset: apre un file esistente in sola lettura, impostando la posizione corrente a inizio file;

· Rewrite: crea e apre un nuovo file in sola scrittura. Se un file esiste già con quel nome viene cancellato;

· Append: apre un file esistente in sola scrittura, impostando la posizione corrente a fine file.

Per chiudere un file occorre chiamare CloseFile passando come unico argomento la variabile di tipo file.

Per leggere da un file di testo si usano le funzioni read e readln che chiedono come primo argomento il file di testo da cui leggere e come argomenti successivi una serie di variabili su cui salvare l’input. Per scrivere su un file di testo si usano invece le funzioni write e writeln che chiedono come primo argomento il file di testo su cui scrivere e come argomenti successivi i valori da scrivere. Se il primo argomento manca viene considerato implicitamente Input per read e readln e Output per write e writeln. Questo è esattamente l’utilizzo che abbiamo fatto di queste procedure in tutti gli esempi visti finora.

Il programma seguente scrive su un file ‘prova.txt’ la scritta “Hello world!”

program TextFileTest;

{$APPTYPE CONSOLE}

uses

  SysUtils;

var F: TextFile;

begin

  AssignFile(F, 'prova.txt');

  Rewrite(F);

  Writeln(F, 'Hello world!');

  CloseFile(F);

end.

Se esiste già un file col nome prova.txt viene cancellato e sovrascritto.

Esistono due funzioni che permettono di capire se si è raggiunta la fine del file o della linea di testo: Eof (End of file) e Eoln (End of line). Vediamo un esempio di come utilizzare queste procedure. Il seguente programma prende l’input da un file di testo ‘Input.txt’ e scrive l’output sul file ‘Output.txt’. In input.txt deve trovare una serie di righe su ciascuna delle quali c’è un intero. Per esempio:

1

4

-2

3

In output scrive la somma dei numeri. Per esempio:

1 + 4 + -2 + 3 = 6

Ecco il listato:

program TextFileTest;

{$APPTYPE CONSOLE}

uses

  SysUtils;

var

  InputF, OutputF: TextFile;

  Totale: integer;

  Val: integer;

  Inizio: boolean;

begin

  writeln('TEST TextFILE 2');

  writeln;

  { Apri i file }

  AssignFile(InputF,'InputFile.txt');

  Reset(InputF);

  AssignFile(OutputF, 'OutputFile.txt');

  Rewrite(OutputF);

  { Inizializza le variabili }

  Totale := 0;

  Inizio := true;

  while not Eof(InputF) do
  begin

    readln(InputF,Val);

    if Inizio then Inizio := false

    else write(OutputF,' + ');

    write(OutputF,Val);

    Totale := Totale + Val;

  end;

  writeln(OutputF,' = ',Totale);

  CloseFile(InputF);

  CloseFile(OutputF);

end.

Il programma ha 3 sezioni: apertura dei file, esecuzione del ciclo principale e chiusura dei file. Il ciclo principale viene eseguito finché non si termina di leggere l’input (while not Eof(inputF) do). In ogni iterazione:

· si legge una riga di input (il primo readln);

· se non è la prima iterazione si scrive un ‘+’ in output;

· si scrive in output l’intero letto in input;

· si aggiorna la variabile totale. 

Alla fine del ciclo viene scritto ‘ = ‘ e il contenuto della variabile Totale.

Analogamente, se l’input fosse tutto su una riga, si potrebbe utilizzare Eoln per testare la fine della stringa. L’utilizzo di eof e eoln ha un inconveniente: se alla fine del file o alla fine della riga mancano solo degli spazi vuoti i test danno comunque esito negativo. Questo nell’esempio sopra comporta che se ci sono righe vuote in fondo al file di input in output compare un ‘+ 0’ di troppo. Per rimediare a questo inconveniente i progettisti di Delphi hanno introdotto due nuove funzioni, SeekEof e SeekEoln, che hanno la stessa funzione di Eof e Eoln ma che ritornano true anche quando alla fine del file o della riga mancano dei blanks. Se nel listato sopra si utilizza SeekEof invece di Eof il problema scompare.

File tipati

Una variabile di tipo typed file si dichiara nel seguente modo:

Var f: file of baseType

Per esempio

Var f: file of integer;

I file tipati sono file che contengono array di dati tutti del tipo baseType. Tipicamente il tipo base è un tipo record. L’utilizzo delle procedure AssignFile e CloseFile è identico a quello dei file di testo: AssignFile associa alla variabile il file esterno, CloseFile chiude il file. Per aprire un file si può utilizzare Reset o Rewrite: Reset apre in I/O un file già esistente, Rewrite apre in I/O un nuovo file e se ne esiste già uno con lo stesso nome lo cancella. Con i file tipati non si può utilizzare Append. L’uso di Reset e Rewrite è un po’ differente da quello dei file di testo: con i file di testo Reset apre un file in sola lettura e Rewrite apre un file in sola scrittura, con i file tipati Reset e Rewrite aprono un file in I/O. In entrambi i casi Reset apre un file già esistente e imposta la posizione corrente all’inizio mentre Rewrite crea un nuovo file.

L’I/O viene fatto sui file tipati tramite l’utilizzo di read e write, mentre l’utilizzo di readln e writeln (leggi riga di testo e scrivi riga di testo) non è permesso e del resto non avrebbe senso. Per entrambe le funzioni il primo argomento è il file, i successivi sono variabili o valori di tipo baseType.

Di seguito vediamo un programma che chiede i dati di un impiegato da tastiera e li scrive su file:

program Impiegati;

{$APPTYPE CONSOLE}

uses

  SysUtils;

const

  NomeFile = 'Impiegati.typ';

type

  TImpiegato = record
    Nome: string[20];

    Cognome: string[40];

    Eta: integer;

  end;

  TImpFile = file of TImpiegato;

procedure ChiediRecord(var imp: TImpiegato);

begin

  write('Inserisci nome: ');

  readln(imp.Nome);

  write('Inserisci cognome: ');

  readln(imp.Cognome);

  write('Inserisci eta'': ');

  readln(imp.Eta);

end;

procedure ScriviRecord(var f: TImpFile; imp: TImpiegato);

begin

  write(f,imp);

end;

var

  Impiegato: TImpiegato;

  MyFile: TImpFile;

begin

  writeln('Programma di esempio per i record');

  writeln;

  AssignFile(MyFile,NomeFile);

  Rewrite(MyFile);

  writeln('Inserisci i dati di un impiegato: ');

  ChiediRecord(Impiegato);

  writeln;

  ScriviRecord(MyFile, Impiegato);

  CloseFile(MyFile);

  readln;

end.

L’utilizzo delle procedure mi permette di puntualizzare una cosa: i file vanno sempre passati per riferimento, mai per valore. Il passaggio per valore di una variabile di tipo file provoca un errore di compilazione
.

Questo programma è un po’ limitato, perché scrive il record su file ma non permette di leggerlo. Inoltre, a causa dell’utilizzo di Rewrite, ogni esecuzione riscriverà il vecchio file. Quello che faremo ora è scrivere un programma che fornisca su console un menu di operazioni tra cui “aggiungere un record” e “visualizza i record inseriti” e che permetta di riaprire il vecchio file se esiste o di creare un file nuovo se non esiste.

Concentriamoci prima su quest’ultimo compito. Il codice per aprire un file esistente o crearne uno nuovo dovrà invocare Reset e se la procedura ritorna un errore provare a utilizzare Rewrite. Reset, quando fallisce, solleva un’eccezione di tipo EinOutError. Potremmo quindi intercettare l’eccezione e gestirla richiamando Rewrite nel modo in cui abbiamo visto nella prima lezione. Comunque, se non si desidera utilizzare le eccezioni, è possibile disabilitare l’utilizzo delle eccezioni nelle chiamate di procedure che svolgono compiti di I/O mediante l’uso della direttiva di compilazione {$I-}. Per leggere l’errore dell’ultima chiamata bisogna allora utilizzare la funzione IoResult che ritorna un codice d’errore o 0 in assenza di errori. IOResult azzera la variabile globale che contiene il codice d’errore dell’ultima chiamata per cui, quando si disabilita le eccezioni, deve essere invocata dopo ogni operazione di I/O. Ecco come apparirà la nostra procedura che apre il file:

procedure OpenImpFile(var f: TImpFile);

begin

  AssignFile(f,NomeFile);

  {$I-}

  Reset(F);

  {$I+}

  if IoResult <> 0 then Rewrite(f);

end;

Poiché le eccezioni vengono subito riabilitate dopo la chiamata a Reset, se Rewrite fallisce sarà sollevata un’eccezione.

Torniamo al corpo principale del programma. Per non complicare troppo il codice inseriamo solo 3 opzioni: “aggiungi record”, “visualizza tutti i record” e “cancella ultimo file”, più un’opzione di uscita. Ecco come deve apparire la parte principale del programma:

var

  MyFile: TImpFile;

  choice: integer;

begin

  writeln('Database Impiegati (esempio file tipati)');

  writeln;

  OpenImpFile(MyFile);

  repeat

    writeln('Menu: ');

    writeln('1 - Aggiungere un record');

    writeln('2 - Visualizzare tutti i record');

    writeln('3 - Cancellare ultimo file');

    writeln('9 - Uscire');

    writeln;

    write('Fai la tua scelta: ');

    readln(choice);

    writeln;

    case choice of
    1: AggiungiRecord(MyFile);

    2: VisualizzaDatabase(MyFile);

    3: TroncaFile(MyFile);

    9: ;

    else writeln('Scelta non valida, riprova')

    end;

    writeln;

  until choice = 9;

  CloseFile(MyFile)

end.

Poiché quando il file viene aperto da OpenImpFile la posizione corrente è a inizio file, abbiamo bisogno di un metodo per posizionarci a piacimento all’interno del file. Questo metodo è dato dalla procedura seek che prende due parametri: il file e la posizione. Per esempio, se f è il file, seek(f,0) si posiziona a inizio file, seek(f,1) dopo il primo elemento, seek(f,2) dopo il secondo elemento e così via. Per leggere la posizione corrente nel file si può utilizzare FilePos(f), mentre FileSize(f) ritorna il numero di elementi attualmente nel file. AggiungiRecord possiamo allora implementarla nella seguente maniera:

procedure AggiungiRecord(var f: TImpFile);

var

  imp: TImpiegato;

begin

  { Chiedi dati }

  writeln('Inserisci i dati di un impiegato');

  ChiediRecord(imp);

  { Modifica file }

  seek(f,FileSize(f));

  write(f, imp);

end;

dove ChiediRecord è una procedura fatta così:

procedure ChiediRecord(var Imp: TImpiegato);

begin

  write('Inserisci nome: ');

  readln(imp.Nome);

  write('Inserisci cognome: ');

  readln(imp.Cognome);

  write('Inserisci eta'': ');

  readln(imp.Eta);

end;

AggiungiRecord chiede da console i dati di un record, dopodiché si posiziona alla fine del file con seek(f,FileSize(f)) e scrive un record. 

L’implementazione di VisualizzaDB è analoga: impostiamo la posizione corrente del file a 0 e verifichiamo che FilePos sia più piccolo di FileSize, in questo modo:

procedure VisualizzaDatabase(var f: TImpFile);

var

  imp: TImpiegato;

begin

  writeln('Inizio database');

  writeln;

  seek(f,0);

  while FilePos(f) < FileSize(f) do
  begin

    read(f,imp);

    writeln('Nome: ', imp.Nome);

    writeln('Cognome: ',imp.Cognome);

    writeln('Eta'': ', imp.Eta);

    writeln;

  end;

  writeln('Fine database (',FileSize(f), ' record trovati)')

end;

Alternativamente è possibile utilizzare la funzione eof che verifica che non si sia ancora raggiunta la fine del file. L’espressione del ciclo while diventa così:

  while not eof(f) do …
Questa soluzione è preferibile anche perché più portabile (FilePos e FileSize sono state introdotte in Dekphi e non esistono in Pascal). Infine, per cancellare l’ultimo record nel file, abbiamo a disposizione la procedura Truncate(f) che tronca il file alla posizione corrente. Ecco l’ultima procedura rimasta:

procedure TroncaFile(var f: TImpFile);

begin

  seek(f,FileSize(f)-1);

  Truncate(f);

end;

Di seguito elenco il listato completo:

program Impiegati;

{$APPTYPE CONSOLE}

uses

  SysUtils;

const

  NomeFile = 'Impiegati.typ';

type

  TImpiegato = record
    Nome: string[20];

    Cognome: string[40];

    Eta: integer;

  end;

  TImpFile = file of TImpiegato;

procedure OpenImpFile(var f: TImpFile);

begin

  AssignFile(f,NomeFile);

  {$I-}

  Reset(F);

  {$I+}

  if IoResult <> 0 then Rewrite(f);

end;

procedure ChiediRecord(var Imp: TImpiegato);

begin

  write('Inserisci nome: ');

  readln(imp.Nome);

  write('Inserisci cognome: ');

  readln(imp.Cognome);

  write('Inserisci eta'': ');

  readln(imp.Eta);

end;

procedure AggiungiRecord(var f: TImpFile);

var

  imp: TImpiegato;

begin

  { Chiedi dati }

  writeln('Inserisci i dati di un impiegato');

  ChiediRecord(imp);

  { Modifica file }

  seek(f,FileSize(f));

  write(f, imp);

end;

procedure TroncaFile(var f: TImpFile);

begin

  seek(f,FileSize(f)-1);

  Truncate(f);

end;

procedure VisualizzaDatabase(var f: TImpFile);

var

  imp: TImpiegato;

begin

  writeln('Inizio database');

  writeln;

  seek(f,0);

  while not eof(f) do
  begin

    read(f,imp);

    writeln('Nome: ', imp.Nome);

    writeln('Cognome: ',imp.Cognome);

    writeln('Eta'': ', imp.Eta);

    writeln;

  end;

  writeln('Fine database (',FileSize(f), ' record trovati)')

end;

var

  MyFile: TImpFile;

  choice: integer;

begin

  writeln('Database Impiegati (esempio file tipati)');

  writeln;

  OpenImpFile(MyFile);

  repeat

    writeln('Menu: ');

    writeln('1 - Aggiungere un record');

    writeln('2 - Visualizzare tutti i record');

    writeln('3 - Cancellare ultimo file');

    writeln('9 - Uscire');

    writeln;

    write('Fai la tua scelta: ');

    readln(choice);

    writeln;

    case choice of
    1: AggiungiRecord(MyFile);

    2: VisualizzaDatabase(MyFile);

    3: TroncaFile(MyFile);

    9: ;

    else writeln('Scelta non valida, riprova')

    end;

    writeln;

  until choice = 9;

  CloseFile(MyFile)

end.

File non tipati

I file non tipati sono molto simili ai file tipati tranne per il fatto che gli elementi di un file non tipato sono blocchi di byte invece che dati di un tipo particolare. Una variabile di file non tipato si dichiara nel seguente modo:

Var f: file;

Le routine AssignFile, CloseFile, Seek, FilePos, FileSize, Eof, Truncate si comportano come per i file tipati, per cui non serve rispiegarle. Per aprire un file si utilizza Reset e Rewrite che hanno lo stesso identico significato che hanno con i file tipati: Reset apre in I/O un file già esistente mentre Rewrite crea apre in I/O un nuovo file. La differenza è che aprendo un file non tipato si può specificare come secondo parametro la dimensione dei blocchi di byte che compongono il file: se la dimensione dei blocchi è omessa si assume che sia di 128 byte.

Per l’input e per l’output non si utilizza Read e Write ma BlockRead e BlockWrite: entrambe le procedure prendono 3 parametri: il file, l’indirizzo di un buffer in cui o da cui copiare i dati e il numero di blocchi che il buffer può ospitare; in più possono prendere un quarto parametro intero passato per riferimento che riempiranno col numero di byte realmente letti o scritti.

Visto che i file non tipati sono i meno usati non li approfondiremo oltre.

Routine di manipolazione dei file esterni

Oltre alle routine che gestiscono file aperti nel programma, ci sono alcune procedure e funzioni che servono per manipolare i file nel file system, che in Pascal prendono il nome di file esterni. Ecco le principali:

· ChDir cambia la directory corrente del programma

· Erase cancella un file 

· GetDir legge la directory corrente di un device

· MkDir crea una directory

· ForceDirectories crea una directory forzando l’intero percorso ad esistere

· Rename rinomina un file

· RmDir cancella una directory non vuota

Unità e incapsulamento dei dati

Concludiamo il corso tornando a parlare delle unit. Le unit servono per spezzare un programma in più file sorgente. Ogni unit inizia con una riga di intestazione che specifica il nome dell’unità ed è composta almeno dalle sezioni interface e implementation, in quest’ordine. La sezione implementation può essere l’ultima della unit o possono esserci le sezioni initialization e finalization (di cui parleremo nel secondo corso).

Tutto ciò che compare nella sezione interface può essere visto dalle altre unit che usano questa unit. Una unit usa altre unit specificandole nella clausola uses. Ciò che compare nella implementation e che non è dichiarato nella interface rimane invece privato alla unit.

Ecco come potremmo spezzare il programma Impiegati che chiede i dati di un impiegato da console e li mostra a video. Il programma principale:

program Impiegati;

{$APPTYPE CONSOLE}

uses

  SysUtils,

  ImpiegatiU in 'ImpiegatiU.pas';

var

  Impiegato: TImpiegato;

begin

  writeln('Programma di esempio per i record');

  writeln;

  writeln('Inserisci i dati di un impiegato: ');

  ChiediRecord(Impiegato);

  writeln;

  writeln('I dati che hai inserito sono: ');

  MostraRecord(Impiegato);

  readln;

end.

E il listato dell’unità:

unit ImpiegatiU;

interface

type

  TImpiegato = record
    Nome: string[20];

    Cognome: string[40];

    Eta: integer;

  end;

procedure ChiediRecord(var imp: TImpiegato);

procedure MostraRecord(imp: TImpiegato);

implementation

procedure ChiediRecord(var imp: TImpiegato);

begin

  write('Inserisci nome: ');

  readln(imp.Nome);

  write('Inserisci cognome: ');

  readln(imp.Cognome);

  write('Inserisci eta'': ');

  readln(imp.Eta);

end;

procedure MostraRecord(imp: TImpiegato);

begin

  writeln('Nome: ', imp.Nome);

  writeln('Cognome: ',imp.Cognome);

  writeln('Eta'': ', imp.Eta);

end;

end.

Si noti che la struttura dati e tutte le procedure che lavorano sulla struttura dati sono state spostate in un’unità. Questa unità potrebbe essere riusata in programmi differenti, permettendo così l’esportazione della struttura dati TImpiegato e di tutto ciò che la riguarda. Questo modo di spezzare il codice (isolare in un’unità una struttura dati e le routine che la utilizzano) prende il nome di incapsulamento dei dati (data encapsulation) ed è l’idea che ha dato origine alla programmazione orientata agli oggetti. Come vedremo nel prossimo corso, mentre nella programmazione classica le procedure (intese come liste di istruzioni) hanno un ruolo centrale, nella programmazione orientata agli oggetti l’attenzione si sposta sulle strutture dati.

� Volendo essere pignoli la procedura è un tipo speciale di subroutine caratterizzata dal fatto di essere identificata da un nome, mentre la subroutine non necessariamente ha un nome (come in Assembler e in Basic) e in questo caso secondo le definizioni che si trovano in letteratura non può essere considerata un modulo


� Anche se la documentazioni dice che il compilatore scarta il valore di una variabile passata con out prima di eseguire le istruzioni nel corpo della procedura, facendo qualche prova ho scoperto che in realtà Delphi 5 non fa nulla del genere


� Nel Pascal di N. Wirth non esisteva il tipo stringa; le stringhe erano memorizzate nei packed array of char


� La documentazione dice che i nuovi valori saranno non definiti, mentre basta fare una rapida verifica per convincersi che Delphi5 azzera la nuova memoria


� Nella terminologia Pascal un file esterno è un file nel file system


� Il motivo è che AssignFile assegna a una variabile di tipo file un record che contiene tutte le informazioni sul file aperto (per esempio la posizione). Un’eventuale passaggio per valore dovrebbe fare una copia di questo record, il ché non ha senso perché le informazioni sul file devono essere univoche





